Includes easy-entry keystroke listings.

HEWLETT-PACKARD

HP-75

USERS' LIBRARY SOLUTIONS

Real Estate

NOTICE

The program material contained herein is supplied without representation or warranty of any kind. Hewlett-Packard Company therefore assumes no responsibility and shall have no liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof.

TABLE OF CONTENTS

PROGRAM DESCRIPTION

INCOME PROPERTY ANALYSIS

The program accepts the mortgage information on an investment, the life of the investment and its income and expenses. Given a user-chosen inflation and appreciation rate, the program then displays the performance of the investment over the life of the project. When all reports have been displayed, the program will display a summary of the after-tax cashflows for each year, the final sale price, and the sales proceeds. The sales commission is an input variable. The program computes the internal rate of return on the investment and displays it.

After the computations have been made, the program will allow the user to review the input data, the final cashflows, sale price, net proceeds and internal rate of return.

SAMPLE PROBLEM

Bob Edwards is interested in purchasing a small apartment complex as an investment for the next three years. The property costs $50,000. Bob will pay 10% down. He will also pay 17.63% on the 35 year mortgage. The tax deductable closing costs are $2,000; the non-deductable closing costs are $1,500. The land is worth $15,000. The total monthly rental income is $2,000 with a vacancy rate of 5%. The annual expenses are: utilities $750; property taxes $1,800; insurance $500; other expenses $150; maintenance is 3% of the rent.

If Bob's marginal tax bracket is 50%, the inflation rate is 13%, the depreciation rate is 10%, the appreciation rate on the property is 22%, and the sales commission is 6%. Is this a good investment if Bob wants an internal rate of return (IRR of 20%)?

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program		
1a	Sign-on message	$ Investment Property Analysis $	
2	Enter term of investment	Years from purchase to sale?	3 [RTN]
	Enter principal of mortgage	Price of property?	50000 [RTN]
	Down payment (as a % of mortgage)	Percent as down payment?	10 [RTN]
	Enter interest rate as percent	Interest rate?	17.63 [RTN]
	Enter tax deductable closing costs	Tax deduct. closing costs?	2000 [RTN]
	Enter non-deductable closing costs	Non-deduct. closing costs?	1500 [RTN]
	Enter term of mortgage	Mortgage term in years?	35 [RTN]
	Enter value of land	Value of the land?	15000 [RTN]
	Enter the monthly rental income	Monthly rental income?	2000 [RTN]
	Enter the vacancy rate	Estimated vacancy rate?	5 [RTN]
	Enter annual utilities	Annual utilities?	750 [RTN]
	Enter annual property tax	Annual property taxes?	1800 [RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Enter the maintenance (as % of rent)	Maint. as (%) of rent?	3 [RTN]
	Enter the annual insurance costs	Annual ins. costs?	500 [RTN]
	Enter the other expenses	Other annual expenses?	150 [RTN]
	Enter the depreciation rate	Depreciation rate (%/yr)?	10 [RTN]
	Enter the marginal tax bracket	Marginal income tax bracket?	50 [RTN]
	Enter the inflation rate	Inflation factor?	13 [RTN]
	Enter the appreciation rate	Appreciation (%) per year?	22 [RTN]
	Enter the sales commission (%)	Sales Commission (%)?	6 [RTN]
3	Display results	Cashflow and income	
		Year 1 cashflows	
		Gross rental income 22800	[RTN]
		Interest expense 7932.03	[RTN]
		Depreciation 3500	[RTN]
		Other expenses 4300	[RTN]
		Total expenses 15732.03	[RTN]
		Net income 7067.97	[RTN]
		Cashflow 10549.16	[RTN]
		After tax cashflow 7015.18	[RTN]
		Year 2 cashflows	
		Gross rental income 25764	[RTN]
		Interest expense 7928.44	[RTN]
		Depreciation 3150	[RTN]
		Other expenses 4859	[RTN]
		Total expenses 15937.44	[RTN]
		Net income 9826.56	[RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
		Cashflow 12954.16	[RTN]
		After tax cashflow 8040.88	[RTN]
		Year 3 cashflows	
		Gross rental income 29113.32	[RTN]
		Interest expense 7924.15	[RTN]
		Depreciation 2835	[RTN]
		Other expenses 5490.67	[RTN]
		Total expenses 16249.82	[RTN]
		Net income 12863.5	[RTN]
		Cashflow 15671.81	[RTN]
		After tax cashflow 9240.06	[RTN]
4	Display summaries	Summary of after-tax cashflow	[RTN]
		Year 0 cashflow -7500	[RTN]
		Year 1 cashflow 7015.18	[RTN]
		Year 2 cashflow 8040.88	[RTN]
		Year 3 cashflow 37841.34	[RTN]
		Final sale price 90792.40	[RTN]
		Proceeds in year 3 = 85344.86	[RTN]
5	State that IRR is being	Internal Rate of Return	
	computed. Display the IRR	At appr. 22%; IRR is 132.74	[RTN]
6	End program.	Run again, View again, or End?	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program		
1a	Sign-on message	$ Investment Property Analysis $	
2	Enter each data item as prompted	Years from purchase to sale?	X [RTN]
	All require an entry, so a	Price of property?	X [RTN]
	[RTN] with no data, will ask	Percent as down payment?	X [RTN]
	for the data again	Interest rate?	X [RTN]
		Tax deduct. closing costs?	X [RTN]
		Non-deduct. closing costs?	X [RTN]
		Mortgage term in years?	X [RTN]
		Value of the land?	X [RTN]
		Monthly rental income?	X [RTN]
		Estimated vacancy rate?	X [RTN]
		Annual utilities?	X [RTN]
		Annual property taxes?	X [RTN]
		Maint. as (%) of rental?	X [RTN]
		Annual ins. costs?	X [RTN]
		Other annual expenses?	X [RTN]
		Depreciation rate (%/yr)?	X [RTN]
		Marginal income tax bracket?	X [RTN]
		Inflation factor?	X [RTN]
		Appreciation (%) per year?	X [RTN]
		Sales Commission (%)?	X [RTN]
3	Display results	Cashflow and income	
3a		Year n cashflow	
	Display annual income and	Gross rental income	[RTN]/[BACK]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
	expense data and then cash-	Interest expense	[RTN]/[BACK]
	flows. A [RTN] advances to	Depreciation	[RTN]/[BACK]
	the next item, [BACK] goes to	Other expenses	[RTN]/[BACK]
	the prior item and [TAB] ends	Total expenses	[RTN]/[BACK]
	the program.	Net income	[RTN]/[BACK]
		Cashflow	[RTN]/[BACK]
		After-tax cashflow	[RTN]/[BACK]
	Repeat step 3a until investment period has been completed		
4	Display summaries for each	Summary of after-tax cashflows	[RTN]
	year until all years have	Year 1 cashflow	[RTN]/[BACK]
	been displayed.		
	Display final sale price:	Final sale price	[RTN]/[BACK]
	Display proceeds:	Proceeds in year n =	[RTN]/[BACK]
5	Compute the IRR	Internal Rate of Return	
5a	IRR is displayed OR	At appr IRR is	[RTN]
5b	Failure message is shown	Unable to compute IRR	[RTN]/[BACK]
6	Display options menu	Run again, View again, or End?	R,V,or E [RTN]
	If 'R' is pressed goto 1a		
	If 'V' is pressed goto 7		
	If 'E' is pressed end routine		
7	View the data - [RTN]	Price of property	[RTN]/[BACK]
	advances to next item, [TAB]	Percent as down pmt	[RTN]/[BACK]
	ends the program, and [BACK]	Interest rate	[RTN]/[BACK]
	goes to prior item or group	Vacancy rate	[RTN]/[BACK]
		Mortgage term in years	[RTN]/[BACK]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
		Value of land	[RTN]/[BACK]
		Monthly rental	[RTN]/[BACK]
		Annual utilities	[RTN]/[BACK]
		Annual property tax	[RTN]/[BACK]
		Maint. as (%) of rent	[RTN]/[BACK]
		Annual ins. costs	[RTN]/[BACK]
		Other expenses	[RTN]/[BACK]
		Depreciation rate (%)	[RTN]/[BACK]
		Marginal tax rate	[RTN]/[BACK]
		Inflation rate	[RTN]/[BACK]
		Appreciation	[RTN]/[BACK]
		Tax deduct. closing	[RTN]/[BACK]
		Non-deduct. closing	[RTN]/[BACK]
		Life of investment	[RTN]/[BACK]
		Sales commission (%)	[RTN]/[BACK]
		Down payment	[RTN]/[BACK]
		Value of buildings	[RTN]/[BACK]
		Annual payment	[RTN]/[BACK]
	Display after tax cashflow: Repeat until all years have	Year n Cashflow n	[RTN]/[BACK]
	been displayed		
	Display final sale price:	Final sale price	[RTN]/[BACK]
	Display proceeds:	Proceeds in year =	[RTN]/[BACK]
	Display IRR:	Internal Rate of Return	[RTN]/[BACK]
	Goto step 6		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
A2	Cashflow in final year of investment	M(15)	Inflation rate
C(0)	Initial cashflow	M(16)	Appreciation
C(1)	Cashflow for year i; 1<i<M(19)	M(17)	Tax deductable closing costs
G1	Final sale price	M(18)	Non-deductable closing costs
I	Index (counter)	M(19)	Lifetime of investment
I6	Compounding factor	M(20)	Sales commission (%)
J	Parameter in rounding routine	M(21)	Balance after down is deducted
J8	Monthly interest rate as a decimal	M(22)	Value of improvements (buildings)
J9	Counter in IRR routine	M(23)	Annual payments
M(1)	Price of the property	M(24)	Value of improvements less depreciation
M(2)	Down payment as a percent of price	M(25)	Balance after down payment is deducted
M(3)	Interest rate	N	Mortgage term in months
M(4)	Estimated vacancy rate	O1	Net annual rental income
M(5)	Mortgage term in years	O2	Interest expenses
M(6)	Value of land	O3	Depreciation in any given year
M(7)	Monthly rental	O4	Other expenses
M(8)	Annual utilities	O5	Total expenses
M(9)	Annual property taxes	O6	Net income
M(10)	Maintenance as a percent of rental price	O7	Marginal tax rate reduction
M(11)	Annual insurance costs	O8	Gross cashflow
M(12)	Other expenses	O9	After tax cashflow
M(13)	Depreciation rate as a percentage	P	Monthly payment
M(14)	Marginal income tax rate	P1	Beginning month in interest computations

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
P2	Ending month in interest computations	Z	Estimate of internal rate of return
R	Compounding rate in interest computation	Z0	Temporary holding value
S2	Monthly payment to interest	Z1	Lower bound on estimate of IRR
S3	Accumulated payment to interest & principal	Z2	Upper bound on estimate of IRR
T	Reduction in final years cashflow	Z3	Prior estimate of IRR
V1	Balance after down is deducted	A$	Contains output labels
X	Parameter in rounding function	K$	Keyboard input
X9	Compounded value used in IRR routine	Q$	Keyboard input - a [RTN] usually follows
Y2	Current year of investment	X$	Alpha input variable, usually converted to numeric

NOTES AND REFERENCES

References:

1. The internal rate of return routine will find values between -99.9% and 1000%. To change the range, change Z1 and Z2.

2. For more about the IRR calculation, refer to the Internal Rate of Return program.

3. The term of the investment may not exceed 30 years.

PROGRAM LISTING

```
 10 ! Given the income and
 20 ! expenses data on a real
 30 ! estate investment,compute
 40 ! the income flows, and both
 50 ! before and after tax cash
 60 ! flows. Also compute the
 70 ! Internal Rate of Return on
 80 ! that investment.
 90 !
100 !
110 ! Revision 11/01/82 - twb-
120 !
130 DELAY 2 @ DIM M(27),C(30),A$[460]
140 !
150 ! Round X to J decimals
160 !
170 DEF FNR(X,J) = INT(X*10^J+.5)/10^J      -FNR rounds X to J decimal
                                             places
180 !
190 ! Single upper-case key in
200 !
210 DEF FNK$                                -FNK$ returns value of the key
                                             pressed by the user
220 K$=KEY$ @ IF K$="" THEN 220
230 FNK$=UPRC$(K$)
240 END DEF
250 !
260 ! Start the program
270 !
280 FOR I=1 TO 441 STEP 20
290 READ A$[I,I+19] @ NEXT I
300 DELAY 5 @ DISP "$ Investment Proper     -Display the sign on message
    ty Analysis $" @ DELAY 2
310 X1=1
320 I6=1
330 Y2=0
340 DISP "Years from purchase to sale";     -Accept input of alpha item
     @ INPUT X$
350 IF X$="" THEN 340                       -If just 'RTN' pressed, ask for
                                             input again
360 IF X$="Q" THEN 3050                     -If user pressed 'Q', then goto
                                             end
370 ON ERROR GOTO 390                       -Set up error routine
380 M(19)=VAL(X$) @ OFF ERROR @ GOTO 40     -Convert to number.
    0
390 DISP "Oops..."; @ GOTO 340              -Display error message and ask
                                             for input again
400 IF M(19)>30 THEN 390                    -If number of years is greater
                                             than limit, error exists
410 DISP "Price of property"; @ INPUT X
    $
420 IF X$="" THEN 410
430 IF X$="Q" THEN 3050
440 ON ERROR GOTO 460
```

PROGRAM LISTING

```
450 M(1)=VAL(X$) @ OFF ERROR @ GOTO 470
460 DISP "Oops..."; @ GOTO 410
470 IF M(1)<=0 THEN 460

480 DISP "Percent as down payment"; @ I
    NPUT X$
490 IF X$="" THEN 480
500 IF X$="Q" THEN 3050
510 ON ERROR GOTO 530
520 M(2)=VAL(X$) @ OFF ERROR @ GOTO 540
530 DISP "Oops..."; @ GOTO 480
540 IF M(2)<=0 THEN 530

550 DISP "Interest rate"; @ INPUT X$
560 IF X$="" THEN 550
570 IF X$="Q" THEN 3050
580 ON ERROR GOTO 600
590 M(3)=VAL(X$) @ OFF ERROR @ GOTO 610
600 DISP "Oops..."; @ GOTO 550
610 IF M(3)<=0 THEN 600

620 DISP "Tax deduct. closing costs"; @
     INPUT X$
630 IF X$="" THEN 620
640 IF X$="Q" THEN 3050
650 ON ERROR GOTO 670
660 M(17)=VAL(X$) @ OFF ERROR @ GOTO 68
    0
670 DISP "Oops..."; @ GOTO 620
680 IF M(17)<0 THEN 670

690 DISP "Non-deduct. closing costs"; @
     INPUT X$
700 IF X$="" THEN 690
710 IF X$="Q" THEN 3050
720 ON ERROR GOTO 740
730 M(18)=VAL(X$) @ GOTO 750
740 DISP "Oops..."; @ GOTO 690
750 IF M(18)<0 THEN 740

760 DISP "Mortgage term in years"; @ IN
    PUT X$
770 IF X$="" THEN 760
780 IF X$="Q" THEN 3050
790 ON ERROR GOTO 810
800 M(5)=VAL(X$) @ OFF ERROR @ GOTO 820
810 DISP "Oops..."; @ GOTO 760
820 IF M(5)<=0 THEN 810

830 DISP "Value of the land"; @ INPUT X
    $
840 IF X$="" THEN 830
850 IF X$="Q" THEN 3050
860 ON ERROR GOTO 880
```

-If negative or zero prop. price, ask for input again

-If percent down is not positive, ask for input again

-If interest rate is not positive ask for input again

-If tax-deduc. closing costs are negative, input again

-If non-deduc. closing costs are negative, input again

-If mortgage term is not positive, input again

PROGRAM LISTING

```
 870 M(6)=VAL(X$) @ OFF ERROR @ GOTO 890
 880 DISP "Oops...";  @ GOTO 830
 890 IF M(6)<0 THEN 880                          -If land value is negative, ask
                                                  for input again
 900 DISP "Monthly rental income"; @ INP
     UT X$
 910 IF X$="" THEN 900
 920 IF X$="Q" THEN 3050
 930 ON ERROR GOTO 950
 940 M(7)=VAL(X$) @ OFF ERROR @ GOTO 960
 950 DISP "Oops...";  @ GOTO 900
 960 IF M(7)<0 THEN 950                          -If rental income is negative,
                                                  ask for input again
 970 DISP "Estimated vacancy rate"; @ IN
     PUT X$
 980 IF X$="" THEN 970
 990 IF X$="Q" THEN 3050
1000 ON ERROR GOTO 1020
1010 M(4)=VAL(X$) @ OFF ERROR @ GOTO 103
     0
1020 DISP "Oops...";  @ GOTO 970
1030 IF M(4)<0 THEN 1020                         -If vacancy rate is negative,
                                                  ask for input again
1040 DISP "Annual utilities"; @ INPUT X$
1050 IF X$="" THEN 1040
1060 IF X$="Q" THEN 3050
1070 ON ERROR GOTO 1090
1080 M(8)=VAL(X$) @ OFF ERROR @ GOTO 110
     0
1090 DISP "Oops...";  @ GOTO 1040
1100 IF M(8)<0 THEN 1090                         -If utilities are negative, ask
                                                  for input again
1110 DISP "Annual property taxes"; @ INP
     UT X$
1120 IF X$="" THEN 1110
1130 IF X$="Q" THEN 3050
1140 ON ERROR GOTO 1160
1150 M(9)=VAL(X$) @ OFF ERROR @ GOTO 117
     0
1160 DISP "Oops...";  @ GOTO 1110
1170 IF M(9)<0 THEN 1160                         -If property taxes are negative
                                                  request input again
1180 DISP "Maint. as % of rental"; @ INP
     UT X$
1190 IF X$="" THEN 1180
1200 IF X$="Q" THEN 3050
1210 ON ERROR GOTO 1230
1220 M(10)=VAL(X$) @ OFF ERROR @ GOTO 12
     40
1230 DISP "Oops..." @ GOTO 1180
1240 IF M(10)<0 THEN 1230                        -If maintenance is negative
                                                  request input again
1250 DISP "Annual ins. costs"; @ INPUT X
     $
```

PROGRAM LISTING

```
1260 IF X$="" THEN 1250
1270 IF X$="Q" THEN 3050
1280 ON ERROR GOTO 1300
1290 M(11)=VAL(X$) @ OFF ERROR @ GOTO 13
     10
1300 DISP "Oops...";  @ GOTO 1250
1310 IF M(11)<0 THEN 1300

1320 DISP "Other annual expenses"; @ INP
     UT X$
1330 IF X$="" THEN 1320
1340 IF X$="Q" THEN 3050
1350 ON ERROR GOTO 1370
1360 M(12)=VAL(X$) @ OFF ERROR @ GOTO 13
     80
1370 DISP "Oops...";  @ GOTO 1320
1380 IF M(12)<0 THEN 1370

1390 DISP "Depreciation rate (%/yr)"; @
     INPUT X$
1400 IF X$="" THEN 1390
1410 IF X$="Q" THEN 3050
1420 ON ERROR GOTO 1440
1430 M(13)=VAL(X$) @ OFF ERROR @ GOTO 14
     50
1440 DISP "Oops...";  @ GOTO 1390
1450 IF M(13)<0 THEN 1440
1460 DISP "Marginal income tax bracket";
      @ INPUT X$
1470 IF X$="" THEN 1460
1480 IF X$="Q" THEN 3050
1490 ON ERROR GOTO 1510
1500 M(14)=VAL(X$) @ OFF ERROR @ GOTO 15
     20
1510 DISP "Oops...";  @ GOTO 1460
1520 IF M(14)<0 THEN 1510

1530 DISP "Inflation factor"; @ INPUT X$
1540 IF X$="" THEN 1530
1550 IF X$="Q" THEN 3050
1560 ON ERROR GOTO 1580
1570 M(15)=VAL(X$) @ OFF ERROR @ GOTO 15
     90
1580 DISP "Oops...";  @ GOTO 1530
1590 IF M(15)<0 THEN 1580

1600 DISP "Appreciation % per year"; @ I
     NPUT X$
1610 IF X$="" THEN 1600
1620 IF X$="Q" THEN 3050
1630 ON ERROR GOTO 1650
1640 M(16)=VAL(X$) @ OFF ERROR @ GOTO 16
     60
1650 DISP "Oops...";  @ GOTO 1600
```

-If insurance is negative request input again

-If other expenses are negative request input again

-If marginal tax bracket is negative, input again

-If the inflation rate is negative, input again

PROGRAM LISTING

```
1660 IF M(16)<0 THEN 1650                    -If appreciation rate is
                                              negative, request input again
1670 DISP "Sales Commission %"; @ INPUT
     X$
1680 IF X$="" THEN 1670
1690 IF X$="Q" THEN 3050
1700 ON ERROR GOTO 1720
1710 M(20)=VAL(X$) @ OFF ERROR @ GOTO 17
     30
1720 DISP "Oops..."; @ GOTO 1670
1730 IF M(20)<0 THEN 1670
1740 DISP "     Cashflow and income"          -Show user that computation is
                                              being done
1750 M(21)=M(1)-M(2)/100*M(1)
1760 M(22)=M(1)-M(6) @ M(24)=M(22)
1770 M(25)=M(21)
1780 GOSUB 2120                              -Compute payment, given term
                                              principal and interest rate
1790 !
1800 ! Display the cashflow for each yea
     r
1810 !
1820 DELAY 1 @ DISP @ DELAY 2 @ Y2=Y2+1      -Increment the year counter
     @ DISP TAB(5);"Year ";Y2;" cashflow
     s"
1830 P1=1+(Y2-1)*12                          -Compute the beginning month
                                              for interest computations
1840 P2=P1+11                                -Compute the number of the
                                              ending month for interest
1850 GOSUB 2260                              -Compute the dollar amount for
                                              interest
1860 O1=12*M(7)*(100-M(4))/100*I6
1870 I6=I6*(1+M(15)/100)
1880 O3=M(24)*M(13)/100
1890 O4=(M(8)+M(9)+M(11)+M(12))*I6+O1*M(
     10)/100
1900 O5=O2+O3+O4
1910 O6=O1-O5
1920 O7=O6*(-M(14)/100)
1930 O8=O1-M(23)-O4
1940 O9=O8+O7
1950 C(Y2)=O9
1960 M(24)=M(24)-O3
1970 M(25)=M(25)-M(23)+O2
1980 !
1990 DISP "Gross rental income";FNR(O1,2     -Display gross rental income,
     ) @ GOSUB 2990 @ IF NUM(Q$)=8 THEN       wait for RTN key
     1990
2000 DISP "Interest expense";FNR(O2,2) @
      GOSUB 2990 @ IF NUM(Q$)=8 THEN 199
     0
2010 DISP "Depreciation";FNR(O3,2) @ GOS
     UB 2990 @ IF NUM(Q$)=8 THEN 2000
2020 DISP "Other expenses";FNR(O4,2) @ G
     OSUB 2990 @ IF NUM(Q$)=8 THEN 2010
```

PROGRAM LISTING

```
2030 DISP "Total expenses";FNR(05,2) @ G
     OSUB 2990 @ IF NUM(Q$)=8 THEN 2020
2040 DISP "Net income";FNR(06,2) @ GOSUB
      2990 @ IF NUM(Q$)=8 THEN 2030
2050 DISP "Cashflow";FNR(08,2) @ GOSUB 2
     990 @ IF NUM(Q$)=8 THEN 2040
2060 DISP "After tax cashflow";FNR(09,2)
      @ GOSUB 2990 @ IF NUM(Q$)=8 THEN 2
     050
2070 IF Y2<M(19) THEN 1820                  -If all years haven't been
                                             displayed, return to start
2080 GOTO 2320                              -Skip subroutines and begin to
                                             compute final data
2090 !
2100 ! Compute the payments
2110 !
2120 S3=M(21)
2130 V1=M(21)
2140 J8=M(3)/1200
2150 N=M(5)*12
2160 Z0=1+J8
2170 R=Z0^(-N)
2180 P=-V1/((1-R)/J8)
2190 P=FNR(P,2)
2200 M(23)=-12*P
2210 RETURN
2220 !
2230 ! Compute the annual
2240 ! interest expenses
2250 !
2260 O2=0
2270 FOR I=P1 TO P2
2280 S2=J8*S3 @ S2=FNR(S2,2) @ S3=S3+S2+
     P @ O2=O2+S2
2290 NEXT I
2300 RETURN
2310 !
2320 C(0)=-M(1)*M(2)/100-M(18)-M(17)*(10
     0-M(14))/100
2330 A2=C(M(19))
2340 G1=M(1)*(1+M(16)/100)^M(19)
2350 S=1-M(20)/100
2360 T=(M(22)-M(24))*M(14)/100+(G1*S-M(1
     ))*.4*M(14)/100
2370 C(M(19))=A2+G1*.94-T-M(25)
2380 !
2390 ! Display summary data
2400 !
2410 DISP "Summary of after-tax cashflow  -If BACK key pressed display
     s" @ GOSUB 2990 @ IF NUM(Q$)=8 THEN     answer again
      2410
2420 FOR I=0 TO M(19)                       -Begin routine to display all
                                             cash flows
2430 C(I)=FNR(C(I),2)
```

PROGRAM LISTING

```
2440 DISP "Year";I;" Cashflow ";C(I) @ G
     OSUB 2990 @ IF NUM(Q$)=13 THEN 2460
2450 I=I-2 @ IF I<1 THEN I=0

2460 NEXT I
2470 DISP "Final sale price";FNR(G1,2) @
      GOSUB 2990 @ IF NUM(Q$)=8 THEN 242
     0
2480 DISP "Proceeds in year";M(19);"=";F
     NR(G1*.94,2) @ GOSUB 2990 @ IF NUM(
     Q$)=8 THEN 2470
2490 !
2500 ! Compute the internal
2510 ! rate of return.
2520 !
2530 DISP "   Internal Rate of Return"
2540 !
2550 Z1=-.999 @ Z2=10

2560 Z3=0 @ J9=0

2570 Z=(Z1+Z2)/2

2580 J9=J9+1 @ IF J9>50 THEN 2790

2590 IF Z=Z3 THEN 2710

2600 Z3=Z
2610 X9=0
2620 FOR J=1 TO M(19)
2630 X9=X9+C(J)/(Z+1)^J

2640 NEXT J
2650 IF X9=-C(0) THEN 2710

2660 IF -C(0)>X9 THEN 2690
2670 Z1=Z
2680 GOTO 2570
2690 Z2=Z
2700 GOTO 2570
2710 DISP "At appr.";M(16);"%; IRR is ";
     FNR(Z*100,2)
2720 GOSUB 2990 @ IF NUM(Q$)=8 THEN 2710

2730 GOTO 3050
2740 !
2750 ! IRR not found after 50
2760 ! tries, so stop and notify
2770 ! User of failure
2780 !
2790 Z=0 @ DISP " Unable to compute IRR"
      @ GOSUB 2990 @ IF NUM(Q$)=8 THEN 2
     790
2800 GOTO 3050
```

-Display year and cash flow. Skip to end if RTN pressed
-Decrease loop counter by 2 to point at previous data
-End of cash flow loop

-For estimate of IRR set initial guess range
-Z3 is the previous guess and J9 is the iteration count
-The first guess is the middle of the range
-If more than 50 guesses, cannot find IRR, exit routine
-If old guess is equal to current one IRR has been found
-Save old guess

-Compute the present value using estimate for interest

-If NPV is equal to investment IRR has been found

-Show the appreciation rate and IRR
-Wait for key input and continue on RTN
-Skip to options menu

PROGRAM LISTING

```
2810 !
2820 ! View the data
2830 !
2840 FOR J=1 TO 23                                  -Begin to show all the input
                                                     data
2850 DISP A$[J*20-19,J*20];" ";M(J) @ GO            -Display label, number and wait
     SUB 2990 @ IF NUM(Q$)=13 THEN 2870              for RTN
2860 J=J-2 @ IF J<1 THEN J=0                        -If BACK was pressed decrement
                                                     index by two
2870 NEXT J
2880 FOR J=0 TO M(19)                               -Begin to display cash flows
                                                     for all years
2890 DISP "Year";J;" Cashflow";C(J) @ GO
     SUB 2990 @ IF NUM(Q$)=13 THEN 2910
2900 J=J-2 @ IF J<1 THEN J=0
2910 NEXT J
2920 DISP "Final Sale Price";FNR(G1,2) @
      GOSUB 2990 @ IF NUM(Q$)=8 THEN 284
     0
2930 DISP "Proceeds in year";M(19);"=";F
     NR(G1*.94,0) @ GOSUB 2990 @ IF NUM(
     Q$)=8 THEN 2920
2940 DISP "Internal Rate of Return";FNR(
     Z*100,3) @ GOSUB 2990 @ IF NUM(Q$)=
     8 THEN GOTO 2930
2950 GOTO 3050
2960 !
2970 ! keyboard input routine
2980 !
2990 Q$=FNK$ @ IF NUM(Q$)#13 AND NUM(Q$)            -Keyboard routine will only
     #8 AND NUM(Q$)#142 THEN 2990                    accept RTN, BACK, or TAB
3000 IF NUM(Q$)=142 THEN 3050                       -If TAB, user wants to end
                                                     program, so goto end
3010 RETURN
3020 !
3030 ! Present options menu
3040 !
3050 DISP CHR$(210);"un again,";CHR$(214
     );"iew again, or ";CHR$(197);
3060 INPUT "nd?"; Q$ @ Q$=UPRC$(Q$)
3070 ON POS('RVE',Q$)+1 GOTO 3050,300,28            -Branch according to which key
     40,3110                                         was pressed
3080 !
3090 ! end of program routine
3100 !
3110 DELAY 1 @ DISP @ STOP                          -Program termination. All stops
                                                     come here.
3120 DATA Price of property,Percent as d
     own pmt,Interest rate,Vacancy rate
3130 DATA Mortgage term in yrs,Value of
     land,Monthly rental,Annual utilitie
     s
3140 DATA Annual property tax,Maint. as
     % of rent,Annual ins costs,Other ex
     penses
```

PROGRAM LISTING

```
3150 DATA Depreciation rate %,Marginal t
     ax rate,Inflation rate,Appreciation
3160 DATA Tax deduct. closing,Non-deduct
     . closing,Life of investment,Sales
     commission %
3170 DATA Down payment,Value of building
     s,Annual payment
```

PROGRAM DESCRIPTION

ESTIMATE OF BUYER'S COSTS

The program acts as a worksheet and computes the down payment, total estimated costs and total estimated cash outlay for a real estate transaction. The program asks for each cost item, and computes the totals at the end of the data entry.

Down payment = Sale price less loan balance

Total estimated costs = Sum of all estimated costs

Total estimated cash outlay = Down payment plus total estimated costs

SAMPLE PROBLEM

Ms. Jones wishes to purchase a house with a selling price of $125,000. She is able to get a loan for $97,000. In addition she must pay a $550 survey fee; a $723 appraisal report; $60 for title insurance; $1,425 tax reserve; escrow fees of $275; attorney fees of $900 and a fire insurance policy of $142.15. If her monthly mortgage payment is $1,100; the annual real estate taxes are $2,793 and the annual property insurance is $201, what is her total monthly payment? ($1,349.50)

What is the down payment? ($28,000)

What are the total estimated costs? ($4,075.15)

What is the total estimated cash outlay? ($32,075.15)

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program		
	Sign-on message	$ Estimate of buyer's costs $	
	Enter sale price	Sale Price?	125000 [RTN]
	Enter loan balance	Loan balance?	97000 [RTN]
	Enter loan fee	Loan fee?	[RTN]
	Enter credit report	Credit report?	[RTN]
	Enter survey fee	Survey costs?	550 [RTN]
	Enter appraisal fee	Appraisal costs?	723 [RTN]
	Enter tax service fee	Tax service fee?	[RTN]
	Enter recording fee	Recording fee?	[RTN]
	Enter interest up to first payment	Interest to 1st pmt?	[RTN]
	Enter title insurance	Mortgagee title ins?	60 [RTN]
	Enter the tax reserve	Tax reserve?	1425 [RTN]
	Enter fire insurance reserve	Fire ins. reserve?	[RTN]
	Enter escrow fees	Escrow fees?	275 [RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Enter tax pro-rates	Tax pro-rates?	[RTN]
	Enter document preparations	Document preparation?	[RTN]
	Enter attorney fees	Attorney fees?	900 [RTN]
	Enter collection escrow fees	Coll. escrow fee?	[RTN]
	Enter pest & dry rot inspection	Pest/dry rot inspec.?	[RTN]
	Enter fire insurance policy	Fire ins. policy?	142.15 [RTN]
	Enter other	Other?	[RTN]
	Enter monthly mortgage	Mortgage payment?	1100 [RTN]
	Enter annual real estate taxes	Annual real est. tax?	2793 [RTN]
	and annual ins. premium	Annual ins. premium?	201 [RTN]
2	Answers	Down payment 28000	[RTN]
		Total est. costs 4075.15	[RTN]
		Est. cash outlay 32075.15	[RTN]
		Mortgage payment 1100	[RTN]
		Monthly taxes 232.75	[RTN]
		Monthly insurance 16.75	[RTN]
		Total Monthly payment 1349.5	[RTN]
3	Review input data:	Run again, View again, or End?	V [RTN]
		Sale price = 125000	[RTN]
		Loan balance = 97000	[RTN]
		Down payment 28000	[RTN]
		Loan fee = 0	[RTN]
		Credit report = 0	[RTN]
		Survey costs = 550	[RTN]
		Appraisal costs = 723	[RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
		Tax service fee = 0	[RTN]
		Recording fee = 0	[RTN]
		Interest to 1st pmt = 0	[RTN]
		Mortgagee title ins. = 60	[RTN]
		Tax reserve = 1425	[RTN]
		Fire ins. reserve = 0	[RTN]
		Escrow fees = 275	[RTN]
		Tax pro-rates = 0	[RTN]
		Document preparation = 0	[RTN]
		Attorney fees = 900	[RTN]
		Coll. escrow fee = 0	[RTN]
		Pest/dry rot inspec. = 0	[RTN]
		Fire ins. policy = 142.15	[RTN]
		Other = 0	[RTN]
		Total est. costs 4075.15	[RTN]
		Est. cash. outlay 32075.15	[RTN]
		Mortgage payment 1100	[RTN]
		Monthly taxes 232.75	[RTN]
		Monthly insurance 16.75	[RTN]
		Total monthly payment 1349.50	[RTN]
	End program	Run again, View again, or End? R	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	$ Estimate of buyer's cost $	
2	Enter data items and press	Sale price?	n or [RTN]
	[RTN] or just press [RTN]	Loan balance?	n or [RTN]
	To quit at this point key	Loan fee?	n or [RTN]
	"Q" (upper-case "Q") and [RTN]	Credit report?	n or [RTN]
		Survey costs?	n or [RTN]
		Appraisal costs?	n or [RTN]
		Tax service fee?	n or [RTN]
		Recording fee?	n or [RTN]
		Interest to 1st pmt?	n or [RTN]
		Mortgagee title ins?	n or [RTN]
		Tax reserve?	n or [RTN]
		Fire ins. reserve?	n or [RTN]
		Escrow fees?	n or [RTN]
		Tax pro-rates?	n or [RTN]
		Document preparation?	n or [RTN]
		Attorney fees?	n or [RTN]
		Coll. escrow fee?	n or [RTN]
		Pest/dry rot inspec.?	n or [RTN]
		Fire ins. policy?	n or [RTN]
		Other?	n or [RTN]
		Mortgage payment?	n or [RTN]
		Annual real est. tax?	n or [RTN]
		Annual ins. premium?	n or [RTN]
3	Display answers	Down payment	[RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
	[TAB] ends the program	Total est. costs	[RTN]/[BACK]
	[BACK] shows the previous item	Est. cash outlays	[RTN]/[BACK]
		Mortgage payment	[RTN]/[BACK]
		Monthly taxes	[RTN]/[BACK]
		Monthly insurance	[RTN]/[BACK]
	[BACK] shows prior group	Total monthly payment	[RTN]/[BACK]
4	Program options:	Run again, View again, or End?	R, V, or E [RTN]
	If 'R' is pressed goto 1		
	If 'V' is pressed:	Sale price =	[RTN]/[BACK]
		Loan balance =	
	[RTN] shows the next item	Down payment =	
		Loan fee =	
	[BACK] shows prior item or	Credit report =	
	prior group	Survey costs =	
		Appraisal costs =	
	[TAB] ends the program	Tax service fee =	
		Recording fee =	
		Interest to 1st pmt. =	
		Mortgagee title ins. =	
		Tax reserve =	
		Fire ins. reserve =	
		Escrow fees =	
		Tax pro-rates =	
		Document preparation =	
		Attorney fees =	

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
		Pest/dry rot inspec. =	
		Coll. escrow fee =	
		Fire ins. policy =	
		Other =	
		Total est. costs =	
		Est. cash outlay =	
		Mortgage payment =	
		Monthly Taxes =	
		Monthly insurance =	
		Total Monthly Payment =	
	Goto 4		
	If 'E' is pressed stop		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
V(1)	Sale price	V(18)	Pest/Dry rot inspection
V(2)	Loan balance	V(19)	Fire insurance policy
V(3)	Loan fee	V(20)	Other
V(4)	Credit report	V(21)	Principal interest and mort. ins.
V(5)	Surveying costs	V(22)	Tax reserves
V(6)	Appraisal fee	V(23)	Insurance reserves
V(7)	Tax service fee	C	Total estimated cash outlay
V(8)	Recording fee	D	Compute down payment
V(9)	Interest (to first payment)	E	Total estimated costs
V(10)	Title ins. for mortgage	F	Parameter in rounding function
V(11)	Tax reserve	I	Index value
V(12)	Fire insurance reserve	J	Precision in rounding function
V(13)	Escrow fee	A$	Strong variable of display labels
V(14)	Tax pro-rates	K$	Single key input
V(15)	Document preparation	Q$	Query response variable
V(16)	Attorney fee	X$	General input variable always converted to numeric
V(17)	Collection escrow fee		

PROGRAM LISTING

```
 10 ! estimate of buyer's costs
 20 !
 30 ! computes down payment,total
 40 ! estimated costs, and total
 50 ! estimated cash outlays for
 60 ! a real estate purchase
 70 !
 80 ! revision 11/01/82
 90 DIM V(23),A$[483]
100 !
110 ! single upper-case key in
120 !
130 DEF FNK$                                  -FNK$ returns a single
                                               upper-case keyboard input
140 K$=KEY$ @ IF K$="" THEN 140
150 FNK$=UPRC$(K$)
160 END DEF
170 !
180 ! round to j digits
190 !
200 DEF FNR(F,J)                              -FNR(I,J) rounds F to J digits
210 F=INT(F*10^J+.5)/10^J
220 FNR=F
230 END DEF
240 !
250 ! data entry loop
260 !
270 DELAY 2
280 DISP " $ Estimate of buyer's cost $
    "
281 FOR I=1 TO 23                             -Read input prompts into A$
282 READ A$[I*21-20,I*21] @ NEXT I
290 FOR I=1 TO 23 @ V(I)=0 @ NEXT I @ D       -Initialize input data
    =0 @ E=0 @ C=0
300 FOR I=1 TO 23                             -Loop to enter input data
310 DISP A$[I*21-20,I*21];
320 INPUT X$ @ IF X$="" THEN 390              -If just RTN pressed, input
                                               again
330 IF X$="Q" THEN 1080                       -If input is 'Q' then terminate
340 ON ERROR GOTO 360
350 V(I)=VAL(X$) @ OFF ERROR @ GOTO 370
360 DISP "Oops..."; @ GOTO 310
370 IF V(I)<0 THEN GOTO 360
380 V(1)=FNR(V(I),2)
390 NEXT I
400 D=V(1)-V(2)                               -Down payment is purchase price
                                               less loan balance
410 FOR I=3 TO 20 @ E=E+V(I) @ NEXT I         -Accumulate expenses
420 C=D+E                                     -Compute total cash outlay
430 V(22)=FNR(V(22)/12,2)
440 V(23)=FNR(V(23)/12,2)
450 !
460 ! Output results
470 !
```

PROGRAM LISTING

```
480 !
490 DISP "Down Payment          ";D          -Display results
500 GOSUB 1020 @ IF NUM(Q$)=8 THEN 490
510 DISP "Total est. costs";E                -Display sign-on message
520 GOSUB 1020 @ IF NUM(Q$)=8 THEN 490
530 DISP "Est. cash outlay";C
540 GOSUB 1020 @ IF NUM(Q$)=8 THEN 510
550 DISP "Mortgage payment      ";V(21)
560 GOSUB 1020 @ IF NUM(Q$)=8 THEN 530
570 DISP "Monthly Taxes         ";V(22)
580 GOSUB 1020 @ IF NUM(Q$)=8 THEN 550
590 DISP "Monthly insurance     ";V(23)
600 GOSUB 1020 @ IF NUM(Q$)=8 THEN 570
610 DISP "Total Monthly Payment";V(21)+
    V(22)+V(23)
620 GOSUB 1020 @ IF NUM(Q$)=8 THEN 590
630 GOTO 960
640 !
650 ! View data
660 !
670 FOR I=1 TO 2
680 DISP A$[I*21-20,I*21];'=';V(I)
690 GOSUB 1020
700 IF NUM(Q$)=13 THEN 720
710 I=I-2 @ IF I<0 THEN I=0
720 NEXT I
730 DISP "Down Payment          ";D
740 GOSUB 1020 @ IF NUM(Q$)=8 THEN 670
750 FOR I=3 TO 20
760 DISP A$[I*21-20,I*21];'=';V(I)
770 GOSUB 1020
780 IF NUM(Q$)=13 THEN 800
790 I=I-2 @ IF I<2 THEN I=2
800 NEXT I
810 DISP "Total est. costs";E
820 GOSUB 1020 @ IF NUM(Q$)=8 THEN 740
830 DISP "Est. cash outlay";C
840 GOSUB 1020 @ IF NUM(Q$)=8 THEN 810
850 DISP "Mortgage payment      ";V(21)
860 GOSUB 1020 @ IF NUM(Q$)=8 THEN 830
870 DISP "Monthly Taxes         ";V(22)
880 GOSUB 1020 @ IF NUM(Q$)=8 THEN 850
890 DISP "Monthly insurance     ";V(23)
900 GOSUB 1020 @ IF NUM(Q$)=8 THEN 870
910 DISP "Total Monthly Payment";V(21)+
    V(22)+V(23)
920 GOSUB 1020 @ IF NUM(Q$)=8 THEN 890
930 !
940 ! present menu
950 !
960 DISP CHR$(210);"un again, ";CHR$(21   -Continuation options
    4);"iew again, or ";CHR$(197);
970 INPUT "nd ","R"; Q$ @ Q$=UPRC$(Q$)
980 ON POS('RVE',Q$)+1 GOTO 960,290,670
    ,1080
```

PROGRAM LISTING

```
 990 !
1000 ! keyboard input subroutine
1010 !
1020 Q$=FNK$ @ IF NUM(Q$)#13 AND NUM(Q$)
     #8 AND NUM(Q$)#142 THEN 1020
1030 IF NUM(Q$)=142 THEN 1080                 -If TAB key was pressed, end
                                               the program
1040 RETURN
1050 !
1060 ! End of the program
1070 !
1080 DELAY 1 @ DISP @ STOP
1090 DATA Sale price,Loan balance,Loan f      -Input prompts
     ee,Credit report,Survey costs
1100 DATA Appraisal costs,Tax service fe
     e,Recording fee,Interest on 1st pmt
1110 DATA Mortgagee title ins,Tax reserv
     e,Fire ins. reserve,Escrow fees,Tax
      pro-rates
1120 DATA Document preparation,Attorney
     fees,Coll. escrow fee,Pest/dry rot
     inspec.
1130 DATA Fire ins. policy,Other,Mortgag
     e payment,Annual real est. tax,Annu
     al ins. premium
```

PROGRAM DESCRIPTION

SELLER'S COSTS AND NET EQUITY

The program acts as a worksheet to compute the total estimated receipts from the sale of real estate. It also computes the total estimated costs and a net equity figure based upon the receipts and costs.

The program will ask for each receipt on cost items and then will present them in summary along with appropriate totals.

Total estimated receipts = sum of the five receipt items.

Total estimated costs = sum of the eighteen expense items.

Net equity = total estimated receipts less total costs.

SAMPLE PROBLEM

Bob Edwards is selling a house for $125,000. His reserve account is $1,196, and his insurance pro-rate is $17.23. If he has to pay $7,500 as a brokerage fee, a first loan balance of $39,351.42, and a $50.00 recording fee, what are his total receipts, costs and net equity?

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program		
	Sign-on message	$ Estimate of Seller's Costs $	
	Enter sale price	Sale price?	125000 [RTN]
	Enter reserve account	Reserve account?	1196 [RTN]
	Skip tax pro-rate	Tax pro-rate 6/30?	[RTN]
	Enter insurance pro-rate	Insurance pro-rate?	17.23 [RTN]
	Skip other receipts	Other receipts?	[RTN]
	Skip title insurance policy	Title ins. policy?	[RTN]
	Skip escrow fee	Escrow fee?	[RTN]
	Enter brokerage fee	Brokerage fee?	7500 [RTN]
	Enter first loan balance	First loan balance?	39351.42 [RTN]
	Skip second loan balance	Second loan balance?	[RTN]
	Skip interest to closing	Interest to closing?	[RTN]
	Skip prepayment penalty	Prepayment penalty?	[RTN]
	Skip tax pro-rate after 7/1	Tax pro-rate 7/1?	[RTN]
	Skip delinquent taxes	Delinquent taxes?	[RTN]
	Skip city or county liens	City or county liens?	[RTN]
	Enter recording fee	Recording costs?	50 [RTN]
	Skip document preparation	Document preparation?	[RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Skip attorney fees	Attorney fees?	[RTN]
	Skip collection escrow fee	Collect. escrow fee?	[RTN]
	Skip loan discount (points)	Loan discount points?	[RTN]
	Skip required repairs	Required repairs?	[RTN]
	Skip judgements	Judgements?	[RTN]
	Skip other costs	Other costs?	[RTN]
2	Answers viewed sequentially	Total Est. Receipts 126213.23	[RTN]
	by pressing [RTN] for the next	Total Est. Costs 46901.42	[RTN]
	answer	Approximate Net Equity 79311.81	[RTN]
	Program options:	Run again, View again, or End? R	V [RTN]
3	View input data and the results	Sale price = 125000	[RTN]
		Reserve account = 1196	[RTN]
		Tax pro-rate 6/30 = 0	[RTN]
		Insurance pro-rate = 17.23	[RTN]
		Other receipts = 0	[RTN]
		Total est. receipts = 126213.23	[RTN]
		Title ins. policy = 0	[RTN]
		Escrow fee = 0	[RTN]
		Brokerage fee = 7500	[RTN]
		First loan balance = 39351.42	[RTN]
		Second loan balance = 0	[RTN]
		Interest to closing = 0	[RTN]
		Prepayment penalty = 0	[RTN]
		Tax pro-rate 7/1 = 0	[RTN]
		Delinquent taxes = 0	[RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
		City or county liens = 0	[RTN]
		Recording costs = 50	[RTN]
		Document preparation = 0	[RTN]
		Attorney fees = 0	[RTN]
		Collect. escrow fee = 0	[RTN]
		Loan discount points = 0	[RTN]
		Required repairs = 0	[RTN]
		Judgements = 0	[RTN]
		Other costs = 0	[RTN]
		Total est. costs = 46901.42	[RTN]
		Approximate net equity 79311.81	[RTN]
	End program	Run again, View again, or End?R	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	$ Estimate of Seller's Costs $	n [RTN]
2	Enter data items and press	Sale price?	n [RTN]
	[RTN] or just press [RTN]. To	Reserve account?	n [RTN]
	quit at this point key "Q"	Tax pro-rate 6/30?	n [RTN]
	(upper-case "Q") and [RTN].	Insurance pro-rate?	n [RTN]
		Other receipts?	n [RTN]
		Title ins. policy?	n [RTN]
		Escrow fee?	n [RTN]
		Brokerage fee?	n [RTN]
		First loan balance?	n [RTN]
		Second loan balance?	n [RTN]
		Interest to closing?	n [RTN]
		Prepayment penalty?	n [RTN]
		Tax pro-rate 7/1?	n [RTN]
		Delinquent taxes?	n [RTN]
		City or county liens?	n [RTN]
		Recording costs?	n [RTN]
		Document preparation?	n [RTN]
		Attorney fee?	n [RTN]
		Collect. escrow fee?	n [RTN]
		Loan discount points?	n [RTN]
		Required repairs?	n [RTN]
		Judgements?	n [RTN]
		Other costs?	n [RTN]
3	Display results	Total Est. Receipts	[RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
		Total Est. Costs	[RTN]/[BACK]
		Approximate Net Equity	[RTN]/[BACK]
4	Display options	Run again, View again, or End? R	R,V, or E [RTN]
	If 'R' then goto 1		
	If 'V' then view input data	Sale price	[RTN]/[BACK]
	an results.	Reserve account	[RTN]/[BACK]
		Tax pro-rate 6/30	[RTN]/[BACK]
		Insurance pro-rate	[RTN]/[BACK]
		Other receipts	[RTN]/[BACK]
		Total est. receipts	[RTN]/[BACK]
		Title ins. policy	[RTN]/[BACK]
		Escrow fee	[RTN]/[BACK]
		Brokerage fee	[RTN]/[BACK]
		First loan balance	[RTN]/[BACK]
		Second loan balance	[RTN]/[BACK]
		Interest to closing	[RTN]/[BACK]
		Prepayment penalty	[RTN]/[BACK]
		Tax pro-rate 7/1	[RTN]/[BACK]
		Delinquent taxes	[RTN]/[BACK]
		City or county liens	[RTN]/[BACK]
		Recording costs	[RTN]/[BACK]
		Document preparation	[RTN]/[BACK]
		Attorney fees	[RTN]/[BACK]
		Collect. escrow fee	[RTN]/[BACK]
		Loan discount points	[RTN]/[BACK]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
		Required repairs	[RTN]/[BACK]
		Judgements	[RTN]/[BACK]
		Other costs	[RTN]/[BACK]
		Total est. receipts	[RTN]/[BACK]
		Total est. costs	[RTN]/[BACK]
		Approximate Net Equity	[RTN]/[BACK]
	Goto 4		
	If 'E' then stop		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
V(1)	Sale Price	V(17)	Document preparation
V(2)	Reserve account	V(18)	Attorney fees
V(3)	Tax pro-rate to 6/30	V(19)	Collection escrow fee
V(4)	Insurance pro-rate	V(20)	Loan discount (points)
V(5)	Other receipts	V(21)	Required repairs
V(6)	Title ins. policy	V(22)	Judgements
V(7)	Escrow fee	V(23)	Other costs
V(8)	Brokerage fee	C	Total estimated costs
V(9)	First loan balance	E	Estimated net equity
V(10)	Second loan balance	F	Parameter in rounding function
V(11)	Interest to closing	I	Index value
V(12)	Prepayment penalty	J	Precision in rounding function
V(13)	Tax pro-rate from 7/1	R	Total estimated receipts
V(14)	Delinquent taxes	A$	String variable of display labels
V(15)	City or county liens	K$	Single key input
V(16)	Recording costs	Q$	Query response variable
		X$	General input variable always converted to numeric

PROGRAM LISTING

```
 10 ! estimate of seller's costs
 20 !
 30 ! computes an estimate of the
 40 ! seller's costs and net equity
 50 ! from a sale of real property
 60 !
 70 ! revision 11/01/82
 80 DIM V(23),A$[506]
 90 ! single upper-case key in
100 !
110 DEF FNK$                                  -FNK$ returns a single
                                               uppercase keyboard input
120 K$=KEY$ @ IF K$="" THEN 120
130 FNK$=UPRC$(K$)
140 END DEF
150 !
160 ! round to J digits
170 !
180 DEF FNR(F,J)                              -FNR rounds F to J digits
190 F=INT(F*10^J+.5)/10^J
200 FNR=F
210 END DEF
220 !
221 RESTORE 890                               -Read item labels into string
                                               array
222 FOR I=1 TO 23
223 READ A$[I*22-21,I*22] @ NEXT I
230 ! data entry loop
240 !
250 DELAY 2
260 DISP " $ Estimate of seller's cost
    $ "
270 FOR I=1 TO 23 @ V(I)=0 @ NEXT I @ C
    =0 @ R=0
280 FOR I=1 TO 23                             -Wait for positive values of
                                               variables
290 DISP A$[I*22-21,I*22];
300 IF V(I) THEN DISP V(I);
310 INPUT X$ @ IF X$="" THEN 380
320 IF X$="Q" THEN 880
330 ON ERROR GOTO 350
340 V(1)=VAL(X$) @ OFF ERROR @ GOTO 360
350 DISP "Oops..."; @ GOTO 290
360 IF V(I)<0 THEN 350
370 V(I)=FNR(V(I),2)
380 NEXT I
390 FOR I=1 TO 5 @ R=R+V(I) @ NEXT I          -Compute the total reciepts
400 FOR I=6 TO 23 @ C=C+V(I) @ NEXT I         -Compute the total costs
410 R=FNR(R,2) @ C=FNR(C,2)                   -Round receipts and costs to
                                               two decimal places
420 E=R-C                                     -Compute net equity
430 !
440 ! output results
450 !
```

PROGRAM LISTING

```
460 !
470 DISP "Total Est. Receipts ";R          -Display results
480 GOSUB 820 @ IF NUM(Q$)=8 THEN 470      -Accept only RTN, BACK or TAB
                                            keys
490 DISP "Total Est. Costs ";C
500 GOSUB 820 @ IF NUM(Q$)=8 THEN 490
510 DISP "Approximate Net Equity";E
520 GOSUB 820 @ IF NUM(Q$)=8 THEN 490
530 GOTO 760
540 !
550 ! View the data
560 !
570 FOR I=1 TO 5 @ DISP A$[1*22-21,I*22
    ];'=';V(I)
580 GOSUB 820
590 IF NUM(Q$)=13 THEN 610
600 I=I-2 @ IF I<0 THEN I=0
610 NEXT I
620 DISP "  Total est. receipts ";R
630 GOSUB 820 @ IF NUM(Q$)=8 THEN 570
640 FOR I=6 TO 23 @ DISP A$[I*22-21,1*2
    2];'=';V(I)
650 GOSUB 820
660 IF NUM(Q$)=13 THEN 680
670 I=1-2 @ IF I<5 THEN I=5
680 NEXT I
690 DISP "  Total est. costs ";C
700 GOSUB 820 @ IF NUM(Q$)=8 THEN 900
710 DISP "Approximate net equity ";E
720 GOSUB 820 @ IF NUM(Q$)=8 THEN 690
730 !
740 ! present menu
750 !
760 DISP CHR$(210);"un again,";CHR$(214
    );"iew again, or ";CHR$(197);
770 INPUT "nd ","R"; Q$ @ Q$=UPRC$(Q$)
780 ON POS('RVE',Q$)+1 GOTO 760,260,570    -Wait for 'R', 'V', or 'E'
    ,880
790 !
800 ! keyboard input subroutine
810 !
820 Q$=FNK$ @ IF NUM(Q$)#13 AND NUM(Q$)
    #8 AND NUM(Q$)#142 THEN 820
830 IF NUM(Q$)=142 THEN 880
840 RETURN
850 !
860 ! end of program routine
870 !
880 DELAY 1 @ DISP @ STOP                  -Program stops here.
890 DATA Sale price,Reserve account,Tax    -Data for item labels
     pro-rate 6/30,Insurance pro-rate
900 DATA Other receipts,Title ins. poli
    cy,Escrow fee,Brokerage fee
910 DATA First loan balance,Second loan
     balance,Interest to closing,Prepay
    ment penalty
```

PROGRAM LISTING

```
920 DATA Tax pro-rate 7/1,Delinquent ta
    xes,City or county liens,Recording
    costs
930 DATA Document preparation,Attorney
    fees,Collect. escrow fee,Loan disco
    unt points
940 DATA Required repairs,Judgements,Ot
    her costs
```

PROGRAM DESCRIPTION

INTERNAL RATE OF RETURN

Given the initial investment, number of periods, and the cashflows for each period, the program will compute the internal rate of return of the cashflows. If given a discount rate and a set of cashflows, the user may compute the net present value of the cashflows. The discount rate is equal to the rate of return which makes the net present value of the investment equal to zero.

The internal rate of return procedure is an iterative search pattern using an interest rate of 50 percent (0.50) as the initial search value. As written, the range of the internal rate of return (IRR) is between -99.9 and 1000 percent. If no solution is found after fifty (50) guesses, then the routine terminates, and places a message into the display that it failed to obtain an accurate estimate of the IRR.

Possible Results

The routine has three possible solutions. The value may be greater than zero; it may be negative; or no solution is found.

Usually no further actions are necessary if the result is positive. If a negative answer is produced, there may be more negative answers and there may be a single positive answer also. Search failure is examined next.

What To Do If the Search Fails?

1. Allow more than fifty passes through the search loop. This can be done by changing the value in the "If" statement to be some number larger than 50.

2. Change the allowable range of interest rates. The current logic uses a range of -0.999 (-99.9%) to 10 (1000%). If you do change these values, choose them so that their sum is not equal to zero.

If a negative result is obtained, a search may be made for a positive answer by altering line 340 to be some value equal to, or larger than, zero. By limiting your search values to positive, if a positive root exists in the range bounded by the values in line 340 and line 360, that positive result should be found.

CAVEAT

Internal Rate of Return computations present some problems since it is possible for a problem to have no solution, or have more than one correct solution. The user should be careful when using IRR to evaluate a situation where the cashflows change sign more than once. Multiple cashflow sign changes are a good indication of possible multiple answers.

SAMPLE PROBLEM

John has a chance to invest $2,000 in a project and he will get $500 for the first two years, $750 in the third and $900 for the last three. If he wants at least a 22% return, should he invest in this project?

(Answer: IRR = 25.06%, so he should invest).

What is the net present value of the investment and cashflows at a 22% discount rate?

(Answer: $170.98).

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program		
	Sign-on message	$ NPV and IRR $	
	Enter initial investment	Investment amount?	2000 [RTN]
	Enter number of years	Number of periods	6 [RTN]
	Skip discount rate	Discount rate (%)	[RTN]
	Enter first year's cashflow	Cashflow for period 1?	500 [RTN]
	Enter second year's cashflow	Cashflow for period 2?	500 [RTN]
	Enter third year's cashflow	Cashflow for period 3?	750 [RTN]
	Enter fourth year's cashflow	Cashflow for period 4?	900 [RTN]
	Enter fifth year's cashflow	Cashflow for period 5?	900 [RTN]
	Enter sixth year's cashflow	Cashflow for period 6?	900 [RTN]
	Select internal rate of return	PRESS Npv OR Irr	I [RTN]
	(pause during computation)	Internal Rate of Return	
	Display answer	Internal Rate of Return = 25.066	[RTN]
	Review data:	Run again, View again, or End? R	V [RTN]
		Number of periods 6	[RTN]
		Initial investment 2000	[RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
		Discount rate 0	[RTN]
		Cashflow in period 1 500	[RTN]
		Cashflow in period 2 500	[RTN]
		Cashflow in period 3 750	[RTN]
		Cashflow in period 4 900	[RTN]
		Cashflow in period 5 900	[RTN]
		Cashflow in period 6 900	[RTN]
		Net present value 0	[RTN]
		Internal rate of retun 25.066	[RTN]
	Options menu - run again	Run again, View again, or End? R	R [RTN]
	Enter investment	Investment amount?	2000 [RTN]
	Enter number of years	Number of periods?	6 [RTN]
	Enter discount rate	Discount rate (%)?	22 [RTN]
	Enter first year's cashflow	Cashflow for period 1?	500 [RTN]
	Enter second year's cashflow	Cashflow for period 2?	500 [RTN]
	Enter third year's cashflow	Cashflow for period 3?	750 [RTN]
	Enter fourth year's cashflow	Cashflow for period 4?	900 [RTN]
	Enter fifth year's cashflow	Cashflow for period 5?	900 [RTN]
	Enter sixth year's cashflow	Cashflow for period 6?	900 [RTN]
	Select net present value	PRESS Npv OR IRR	N [RTN]
	(pause during computation)	Net Present Value	
	Display answer	Net Present Value = 171.01	[RTN]
	Options menu - end program	Run again, View again, or End? R	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	$ NPV and IRR $	
2	Enter data items and press	Investment amount?	n [RTN]
	[RTN] or just press [RTN].	Number of periods?	n [RTN]
	To quit at this point key "Q"	Discount rate (%)?	n [RTN]
	(upper-case "Q") and [RTN]	Cashflow for period 1?	n [RTN]
	Repeat for all periods.	Cashflow for period n?	n [RTN]
3	Select either internal rate	PRESS Npv OR Irr	n or I [RTN]
	of return or net present		
	value. If 'N' goto 4		
	(pause during computation)	Internal Rate of Return	
	Answer displayed	Internal Rate of Return	[RTN]/[BACK]
	Goto 5		
4	(pause during computation)	Net present value	
	Answer displayed	Net present value =	[RTN]/[BACK]
5	Options menu presented	Run again, View again, or End? R	R,V, or E [RTN]
6	If View is selected, the data	Number of periods	[RTN]
	items are presented one at a	Initial investment	[RTN]/[BACK]
	time. To see the next one	Discount rate	[RTN]/[BACK]
	press [RTN]. To back up to	Cashflow in period 1	[RTN]/[BACK]
	prior answer or prior group	Net present value	[RTN]/[BACK]
	press [BACK]. To end the	Internal Rate of Return	[RTN]/[BACK]
	program, press [TAB].		
	Goto 5.		
7	If End is selected then stop.		
8	If Run is selected goto 1		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
C( )	Cashflows in each year	T	Total of all compounded cashflows used to compute the estimate of R
CO	Initial investment		
D	Discount rate	X	Number to be rounded
J	Index value in for next loop	Y	Precision of digits in rounding function
N	Number of years	K$	Single character keyed in and returned as FNK$
P	Net present value	Q$	Keyboard response
R	Internal rate of return	X$	General input, usually converted to numeric. Must be followed by [RTN].

NOTES AND REFERENCES

Reference: PRACTICAL BASIC PROGRAMS, Poole, Lon, Editor, (Osborne/McGraw-Hill, 1980).

PROGRAM LISTING

```
 10 ! Compute the Net Present Value
 20 ! or the Internal Rate of
 30 ! Return on an investment
 40 !
 50 ! revision 11/01/82 -- twb --
 60 !
 70 DELAY 2
 80 DISP TAB(10);"$ NPV and IRR $"
 90 !
100 ! Round X to J decimals
110 !
120 DEF FNR(X,J) = INT(X*10^J+.5)/10^J
130 !
140 ! Single upper-case key in
150 !
160 DEF FNK$                              -FNK$ returns a single
                                           uppercase keyboard input
170 K$=KEY$ @ IF K$="" THEN 170
180 FNK$=UPRC$(K$)
190 END DEF
200 DIM C(100)
210 GOSUB 650                             -Get input data
220 DISP "PRESS ";CHR$(206);"pv OR ";CH   -Select IRR or NPV
    R$(201);"rr "; @ INPUT "";K$
230 IF UPRC$(K$)="N" THEN 250
240 IF UPRC$(K$)="I" THEN 330 ELSE 220
250 DISP TAB(10);"Net Present Value"
260 P=-C0
270 ! Add Present Values for each year
    based on rate of D
280 FOR J=1 TO N
290 P=P+FNR(C(J)/D^J,2)
300 NEXT J
310 DISP "Net Present Value = ";P @ GOS   -Display NPV and goto options
    UB 1160 @ IF NUM(Q$)=8 THEN 310        menu
320 GOTO 1220
330 DISP TAB(5);"Internal Rate of Retur
    n"
340 L=-.999
350 I=0
360 H=10
370 R0=0
380 R=(L+H)/2
390 I=I+1 @ IF I>50 THEN 600              -Allow no more than 50
                                           iterations
400 IF R=R0 THEN 520
410 R0=R
420 T=0
430 FOR J=1 TO N
440 T=T+C(J)/(R+1)^J
450 NEXT J
460 IF T=C0 THEN 520                      -If total = initial investment
                                           IRR has been found
470 IF C0>T THEN 500                      -Modify range of search
```

PROGRAM LISTING

```
480 L=R
490 GOTO 380
500 H=R
510 GOTO 380
520 R=FNR(R*100,3)
530 DISP "Internal Rate of Return=";R @
     GOSUB 1160 @ IF NUM(Q$)=8 THEN 530
540 GOTO 1220
550 !
560 ! IRR not found after 30
570 ! tries, so stop and notify
580 ! User of failure
590 !
600 DISP "IRR is out of current range"
    @ GOSUB 1160 @ IF NUM(Q$)=8 THEN 60
    0
610 R=0 @ GOTO 1220
620 !
630 ! Get the data
640 !
650 C0=0 @ N=0 @ R=0 @ D=1 @ T=0 @ P=0
660 FOR J=1 TO 100 @ C(J)=0 @ NEXT J
670 DISP "Investment amount? ";
680 INPUT ""; X$ @ IF X$="" THEN 670
690 IF X$="Q" THEN 1280

700 ON ERROR GOTO 720

710 C0=VAL(X$) @ OFF ERROR @ GOTO 730
720 DISP "Oops..."; @ GOTO 670
730 IF C0<0 THEN 720

740 C0=FNR(C0,2)
750 DISP "Number of periods?";
760 INPUT ""; X$ @ IF X$="" THEN 760
770 IF X$="Q" THEN 1280
780 ON ERROR GOTO 800
790 N=VAL(X$) @ OFF ERROR @ GOTO 810
800 DISP "Oops..."; @ GOTO 750
810 IF N<0 THEN 800
820 DISP "Discount rate (%)?";
830 INPUT ""; X$ @ IF X$="" THEN 900
840 IF X$="Q" THEN 1280
850 ON ERROR GOTO 870
860 D=VAL(X$) @ OFF ERROR @ GOTO 880
870 DISP "Oops..."; @ GOTO 820
880 IF D<0 THEN 870
890 D=D/100+1
900 FOR J=1 TO N
910 DISP "Cash flow for period ";J;"? "
    ;
920 INPUT ""; X$ @ IF X$="" THEN 910
930 IF X$="Q" THEN 1280
940 ON ERROR GOTO 960
```

-If input is null ask again
-If user enters 'Q' terminate program
-Set error trap for conversion to numeric

-If investment amount is negative ask again

PROGRAM LISTING

```
 950 C(J)=VAL(X$) @ OFF ERROR @ GOTO 970
 960 DISP "Oops...";  @ GOTO 910
 970 NEXT J
 980 RETURN                                        -End of data entry
 990 !
1000 ! View data
1010 !
1020 DISP "Number of periods ";N @ GOSUB
      1160 @ IF NUM(Q$)=8 THEN 1020
1030 DISP "Initial investment ";C0 @ GOS
     UB 1160 @ IF NUM(Q$)=8 THEN 1020
1040 DISP "Discount rate ";(D-1)*100 @ G
     OSUB 1160 @ IF NUM(Q$)=8 THEN 1030
1050 FOR J=1 TO N
1060 DISP "Cash flow in period ";J;" ";C
     (J)
1070 GOSUB 1160 @ IF NUM(Q$)=13 THEN 109
     0
1080 J=J-2 @ IF J<0 THEN J=0
1090 NEXT J
1100 DISP "Net Present Value ";P @ GOSUB
      1160 @ IF NUM(Q$)=8 THEN 1040
1110 DISP "Internal Rate of Return ";R @
      GOSUB 1160 @ IF NUM(Q$)=8 THEN 110
     0
1120 GOTO 1220
1130 !
1140 ! Keyboard input routine
1150 !
1160 Q$=FNK$ @ IF NUM(Q$)#13 AND NUM(Q$)      -Accept only RTN, BACK or TAB
     #8 AND NUM(Q$)#142 THEN 1160                 keys
1170 IF NUM(Q$)=142 THEN 1280
1180 RETURN
1190 !
1200 ! Present options menu
1210 !
1220 DISP CHR$(210);"un again, ";CHR$(21
     4);"iew again or ";CHR$(197);
1230 INPUT "nd?","R"; Q$ @ Q$=UPRC$(Q$)
1240 ON POS('RVE',Q$)+1 GOTO 1220,210,10
     20,1260
1250 !
1260 ! End of program routine
1270 !
1280 DELAY 1 @ DISP @ STOP
```

PROGRAM DESCRIPTION

RENT VS. BUY

The question of whether to rent or to purchase a residence is not always easy to answer, especially when the time period over which you would own or rent a house is short. This program performs an analysis which could be helpful in reaching a decision. The total costs of owning a house and the gains from the sale of the house are compared against the costs of renting a house for the same period of time and investing the money not applied to house purchase in a savings account or other investment opportunity. The program takes into account the tax advantages on property taxes and mortgage interest.

SAMPLE PROBLEM

Bill is considering buying a house for $75,000 at 15% for 25 years. He has $15,000 for a down payment. He can rent the residence for $4,200 per year. As a renter he will have to pay $600 per year for utilities. He expects his rent and expenses to increase at 10% per year. If he purchases the house he will pay $1,100 per year property taxes, $800 per year for utilities and services; the property will appreciate at 10% and his expenses will increase at 10% per year. He plans on living in the house for 5 years. He has a closing cost of $1,200, a marginal tax rate of 40%, has a savings account at 8% and will pay a commission of 6% at sale. What is his gain from the sale; and the ownership advantage?

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program		
	Sign on message	$ Rent vs Buy $	
	Enter the total mortgage	Home purchase cost?	75000 [RTN]
	Enter the down payment	Down payment?	15000 [RTN]
	Enter the closing costs	Closing costs?	1200 [RTN]
	Enter the mortgage interest rate	Mortgage interest (%)?	15 [RTN]
	Enter the term of the mortgage	Mortgage term?	25 [RTN]
	Enter the annual property taxes	Annual property tax?	1100 [RTN]
	Enter the annual utilities and expenses	Annual expenses?	800 [RTN]
	Enter the appreciation rate	Annual appreciation (%)?	10 [RTN]
	Enter the growth rate of expenses	Annual change in exp (%)?	10 [RTN]
	Enter the marginal tax rate	Marginal tax rate (%)?	40 [RTN]
	Enter the interest rate on savings	Savings account rate (%)?	8 [RTN]
	Enter the sales commission	Sales commission (%)?	6 [RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
	What is annual rent?	Annual rent?	4200 [RTN]
	What are renters utilities?	Annual rent expense?	600 [RTN]
	How fast does the rent increase?	Annual change in rent (%)?	10 [RTN]
	Enter annual rate of change in expenses?	Annual change in exp (%)?	10 [RTN]
	What is the life of the investment?	Hold period (years)?	5 [RTN]
2	Results: Total home cost:	Total home cost: 41796.65	[RTN]
	Total rent cost:	Total rent cost: 29304.48	[RTN]
	Gain from sale:	Gain from sale: 38979.44	[RTN]
	Advantage of ownership:	Ownership advantage: 26487.27	[RTN]
	End of program	Run again, View again, or End?	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program		
1a	Sign-on message	$ Rent vs Buy $	
2	Enter the data	Home purchase cost?	X [RTN]
		Down payment?	X [RTN]
		Closing costs?	X [RTN]
		Mortgage interest (%)?	X [RTN]
		Mortgage term?	X [RTN]
		Annual property tax?	X [RTN]
		Expenses?	X [RTN]
		Annual appreciation (%)?	X [RTN]
		Annual change in exp (%)?	X [RTN]
		Marginal tax rate (%)?	X [RTN]
		Savings account rate (%)?	X [RTN]
		Sales commission (%)?	X [RTN]
		Yearly rent?	X [RTN]
		Annual rent expense?	X [RTN]
		Annual change in rent (%)?	X [RTN]
		Annual change in exp (%)?	X [RTN]
		Hold period (years)?	X [RTN]
3	Results: Total home costs:	Total home cost:	[RTN]/[BACK]
	Total rent costs:	Total rent cost:	[RTN]/[BACK]
	Gain from sale:	Gain from sale:	[RTN]/[BACK]
	Advantage of ownership:	Ownership advantage	
4	Present options menu	Run again, View again, or End?	R,V, or E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
	If 'R' is pressed, goto step 1a		
	If 'V' is pressed, goto step 5		
	If 'E' is pressed, then end		
5	Each item is presented for	Home purchase cost	[RTN]/[BACK]
	review	Down payment	[RTN]/[BACK]
		Closing costs	[RTN]/[BACK]
	Press [RTN] to see the next	Mortgage interests (%)	[RTN]/[BACK]
	item	Mortgage term	[RTN]/[BACK]
	Press [BACK] to see the prior	Annual property tax	[RTN]/[BACK]
	item	Annual expenses	[RTN]/[BACK]
	Press [TAB] to terminate the	Annual appreciation (%)	[RTN]/[BACK]
	program	Annual change in exp (%)	[RTN]/[BACK]
		Marginal tax rate (%)	[RTN]/[BACK]
		Savings account rate (%)	[RTN]/[BACK]
		Sales commission (%)	[RTN]/[BACK]
		Annual rent	[RTN]/[BACK]
		Annual rent expense	[RTN]/[BACK]
		Annual change in rent (%)	[RTN]/[BACK]
		Annual change in exp (%)	[RTN]/[BACK]
		Holding period (years)	[RTN]/[BACK]
		Total home cost:	[RTN]/[BACK]
		Total rent cost:	[RTN]/[BACK]
		Gain from sale:	[RTN]/[BACK]
		Ownership advantage:	[RTN]/[BACK]
	Goto step 4		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
A(1)	Home purchase cost	C(3)	Current balance of the loan
A(2)	Down payment in dollars	C(4)	Monthly payment in the loan
A(3)	Closing costs in dollars	C(5)	Accumulative expenses of ownership
A(4)	Mortgage interest rate, percentage	C(6)	Accumulative property taxes
A(5)	Term of mortgage in years	C(7)	Interest lost on down payment and closing costs
A(6)	Yearly property tax expenses	C(8)	Income tax savings
A(7)	Yearly other expenses	C(9)	Total rent paid
A(8)	Yearly appreciation rate, percentage	C(10)	Total expenses paid as renter
A(9)	Yearly expense growth rate, percentage	C(11)	Value of property at time of sale
A(10)	Mortgage tax rate, percentage	I	Index value
A(11)	Savings account interest rate, percentage	I1	Interest on loan in dollars
A(12)	Sales commission, percentage	I9	Effective term of investment
A(13)	Yearly rental payments	P	Parameter in rounding function
A(14)	Yearly rent expenses	P1	Payment to principal
A(15)	Yearly rent growth rate, percentage	T1	Temporary value used in computations
A(16)	Yearly rent expense growth rate, percentage	T2	Temporary value used in computations
A(17)	Holding period in years	T3	Temporary value used in computations
B(1)	Total home ownership costs	X	Parameter in rounding function
B(2)	Total costs of renting	K$	Keyboard response
B(3)	Gain from sale	Q$	Value of key input
B(4)	Ownership advantage	X$	Alpha value, usually converted to numeric
C(1)	Accumulative payment to interest		
C(2)	Accumulative payment to principal		

NOTES AND REFERENCES

Notes:

1. No capital gains taxes are paid at the time of sale.
2. The interest lost in down payment + closing costs is considered a cost of home ownership. i.e. (Down payment + closing) * savings account interest rate = dollars lost.
3. Mortgage interest paid is a deductable item and savings (after tax) is computed using marginal tax bracket.
4. Property tax paid is a deductable item and after tax savings is computed using marginal tax bracket.
5. Some tax savings are realized by not paying taxes on the interest that would have been earned on the down payment and closing.
6. The total home ownership tax savings is the sum of items #3, #4, and #5.
7. The home loan is assumed to be an amortized loan paid monthly.
8. Property taxes increase at "yearly appreciation rate".

PROGRAM LISTING

```
 10 ! Compute the costs of
 20 ! buying a house, and the
 30 ! costs of renting a house,
 40 ! and compare the
 50 ! advantages of ownership or
 60 ! renting
 70 !
 80 ! Revision 11/01/82 - twb -
 90 !
100 DELAY 2
110 !
120 ! Round X to P decimals
130 !
140 DEF FNR(X,P) = INT(X*10^P+.5)/10^P    -FNR rounds X to P digits
150 !
160 ! Single upper-case key in
170 !
180 DEF FNK$                              -FNK$ moniters the keyboard and
                                           returns the first key
190 K$=KEY$ @ IF K$="" THEN 190
200 FNK$=UPRC$(K$)
210 END DEF
220 DIM A(17),B(4),C(11),B$[80],A$[425]
221 FOR I=1 TO 17
222 READ A$[I*25-24,I*25] @ NEXT I
230 FOR I=1 TO 4
240 READ B$[I*20-19,I*20] @ NEXT I
250 !
260 ! Input the data
270 !
280 DISP "        $ Rent vs Buy $"        -Display sign on message
290 RESTORE 1030 @ FOR I=1 TO 17
300 DISP A$[I*25-24,I*25]; @ INPUT X$ @   -Display current label, accept
     ON ERROR GOTO 340                     input and set error trap
310 IF X$="" THEN 300                     -If null input repeat input
                                           prompt
320 IF X$="Q" THEN 960                    -If 'Q' terminate program
330 A(I)=VAL(X$) @ OFF ERROR @ GOTO 350   -Convert alpha input to numeric
340 DISP "Oops..."; @ GOTO 300
350 NEXT I
360 DISP
370 FOR I=1 TO 11 @ C(I)=0 @ NEXT I       -Initialize all accumulators to
                                           zero
380 T1=1/(1+A(4)/1200) @ T2=A(4)/1200 @
     T3=A(1)-A(2)
390 C(4)=T3*T2/(1-T1^(12*A(5)))           -Compute the loan payment
400 C(3)=A(1)-A(2)
410 FOR I=1 TO 12*A(17)
420 I1=C(3)*A(4)/1200
430 P1=C(4)-I1
440 C(3)=C(3)-P1                          -Compute the current loan
                                           balance
450 C(1)=C(1)+I1                          -Accumulate the interest paid
                                           on the loan
```

PROGRAM LISTING

```
460 C(2)=C(2)+P1                                  -Accumulate the principal paid
                                                   on the loan
470 NEXT I
480 FOR I=1 TO A(17)
490 I9=I-1
500 C(5)=C(5)+A(7)*(1+A(9)/100)^I9                -Accumulate the expenses of
                                                   home ownership
510 NEXT I
520 FOR I=1 TO A(17)
530 I9=I-1
540 C(6)=C(6)+A(6)*(1+A(8)/100)^I9                -Accumulate property taxes paid
550 NEXT I
560 T1=A(2)+A(3) @ T2=1+A(11)/100
570 C(7)=T1*T2^A(17)-T1                           -Compute the interest cost when
                                                   down and closing paid
580 C(8)=(C(1)+C(6)+C(7))*A(10)/100               -Compute the income tax savings
590 FOR I=1 TO A(17)
600 I9=I-1
610 C(9)=C(9)+A(13)*(1+A(15)/100)^I9              -Accumulate paid rent
620 C(10)=C(10)+A(14)*(1+A(16)/100)^I9            -Accumulate total renter's
                                                   expenses
630 NEXT I
640 B(1)=C(1)+C(2)+C(5)+C(6)+C(7)-C(8)            -Compute total costs of owning
                                                   the home
650 B(2)=C(9)+C(10)                               -Compute the total cost of
                                                   renting
660 C(11)=A(1)*(1+A(8)/100)^A(17)                 -Compute price of property at
                                                   time of sale
670 B(3)=C(11)-C(11)*A(12)/100-C(3)-A(2
    )-A(3)                                        -Compute the gain from the sale
680 B(4)=B(2)+B(3)-B(1)                           -Compute the advantage of
                                                   ownership
690 !
700 ! View the results
710 !
720 FOR I=1 TO 4
730 DISP B$[I*20-19,I*20];FNR(B(I),2) @
     GOSUB 900                                    -Display the label, item, and
                                                   wait for keyboard
740 IF NUM(Q$)=13 THEN 760                        -If RTN pressed, skip to end of
                                                   loop
750 I=I-2 @ IF I<1 THEN I=0                       -Decrement by 2 for BACK key
760 NEXT I
770 GOTO 960
780 !
790 ! View the data
800 !
810 FOR I=1 TO 17
820 DISP A$[I*25-24,I*25];FNR(A(I),2) @
     GOSUB 900
830 IF NUM(Q$)=13 THEN 850
840 I=I-2 @ IF I<1 THEN I=0
850 NEXT I
860 GOTO 720
870 !
```

PROGRAM LISTING

```
 880 ! Keyboard input routine
 890 !
 900 Q$=FNK$ @ IF NUM(Q$)#13 AND NUM(Q$)
     #8 AND NUM(Q$)#142 THEN 900
 910 IF NUM(Q$)=142 THEN 960

 920 RETURN
 930 !
 940 ! Present options menu
 950 !
 960 DISP CHR$(210);"un again, ";CHR$(21
     4);"iew again, or ";CHR$(197);
 970 INPUT "nd?"; Q$ @ Q$=UPRC$(Q$)
 980 ON POS('RVE',Q$)+1 GOTO 960,280,810
     ,1020
 990 !
1000 ! End of program routine
1010 !
1020 DELAY 1 @ DISP @ STOP
1030 DATA Home purchase cost,Down paymen
     t,Closing costs,Mortgage interest (
     %)
1040 DATA Mortgage term,Annual property
     tax,Annual expenses
1050 DATA Annual appreciation (%),Annual
      change in exp (%),Marginal tax rat
     e (%)
1060 DATA Savings account rate (%),Sales
      commission (%),Annual rent,Annual
     rent expense
1070 DATA Annual change in rent(%),Annua
     l change in exp (%),Holding period
     (years)
1080 DATA Total home cost:,Total rent co
     st:,Gain from sale:,Ownership advan
     tage:
```

-Moniter keyboard for RTN, BACK, or TAB keys
-If TAB key pressed, terminate program

-Display options with initial letters underlined

-Accept only R, V, or E

-Program termination.

PROGRAM DESCRIPTION

VARIABLE PAYMENT MORTGAGE AMORTIZATION TABLES

Given the principal and interest rate on a loan and the number of different monthly payments, the program will accept payment number and monthly payment amount. The program will then generate a schedule of monthly payments showing the amount applied against principal and the amount paid as interest. The amortization table may be printed* if the user wishes.

After the table has been generated, the user may elect to run the program with new data, view the data that was entered, or to quit.

At any time the user is viewing the amortization schedule, by entering "Q" (upper-case "Q") the program will terminate.

* for a printed table, a printer must be attached and defined as device ":PR".

SAMPLE PROBLEM

Barbara is buying land for $5,000. The interest rate on the mortgage is 17.53%. The mortgage begins with a monthly payment of $350. Beginning in month 14 the payments become $500 per month. What is the repayment schedule? The first payment begins on June 10, 1983.

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program		
1a	Sign-on message	$ Mortgage Amortization $	
	Enter the principal	Principal?	5000 [RTN]
	Enter the interest rate	Annual interest rate (%)?	17.53 [RTN]
	Enter the number of different payments	Number of different payments?	2 [RTN]
	Enter payment number and amount	Payment #, payment $?	1,350 [RTN]
	Enter payment number and amount	Payment #, payment $?	14,500 [RTN]
	Enter the beginning date	Beginning date (M,D,Y)?	6,10,1983 [RTN]
	Select no printed output	Do you wish a printout?	N [RTN]
2	Produce amortization table:	Amortization Table	
	Display month, payment to the	6/10/1983 P=276.96 I=73.04	[RTN]
	principal and payment on the	7/10/1983 P=281 I=69	[RTN]
	interest	8/10/1983 P=285.11 I=64.89	[RTN]
		9/10/1983 P=289.27 I=60.73	[RTN]
		10/10/1983 P=293.5 I=56.5	[RTN]
		11/10/1983 P=297.79 I=52.21	[RTN]
		12/10/1983 P=302.14 I=47.86	[RTN]
3	Display year-to-date totals:	1983 end P=2025.77 I=424.23	[RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
		1/10/1984 P=306.55 I=43.45	[RTN]
		2/10/1984 P=311.03 I=38.97	[RTN]
		3/10/1984 P=315.57 I=34.43	[RTN]
		4/10/1984 P=320/18 I=29.82	[RTN]
		5/10/1984 P=324.86 I=25.14	[RTN]
		6/10/1984 P=329.61 I=20.39	[RTN]
		7/10/1984 P=484.42 I=15.58	[RTN]
		8/10/1984 P=491.50 I=8.50	[RTN]
		9/10/1984 P=90.50 I=1.32	[RTN]
		1984 end P=2974.23 I=217.6	[RTN]
4	Options menu	Run again, View again, or End?	V [RTN]
5	View input data	Principal is 5000	[RTN]
		Annual interest (%) 17.53	[RTN]
		Monthly payment 1 = 350	[RTN]
		Monthly payment 2 = 500	[RTN]
6	End program	Run again, View again, or End?	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	$ Mortgage Amortization $	
2	Enter each data item as prompt-	Principal?	n [RTN]
	ed. It "Q" (upper-case "Q")	Annual interest rate?	i [RTN]
	is entered, then program will	Number of different payments?	# [RTN]
	quit.	Payment #, payment $?	m,p [RTN]
		(repeat until done)	
		Beginning date (M,D,Y)?	m,d,y [RTN]
3	Determine if printout is wanted	Do you wish a printout? (Y/N)	Y or N [RTN]
4a	If printout not wanted:	Amortization Table	
	To advance to next line, press	mm/dd/yyyy P=*value* I=*value*	[RTN]/[BACK]
	[RTN]. To exit the program,	(repeats until year end)	
	press [TAB]. [BACK] repeats	yyyy end P=*value* I=*value*	[RTN]/[BACK]
	the current line.	(repeat monthly outputs)	
4b	If printout is wanted:	Amortization Table	
	This is printed	mm/dd/yyyy	
	for each year:	P=*value* I=*value*	
		(repeats until year end)	
		yyyy end	
		P=*value* I=*value*	
		(repeat monthly output)	
5	Display options menu	Run again, View again, or End?	R,V or E [RTN]
	If 'V' then: [RTN] for next	Principal is	[RTN]/[BACK]
	item. [BACK] for prior one.	Annual interest (%)	
	[TAB] ends routine. When done	Monthly Payment n = value	
	goto 5. If 'E' then stop.		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
D	Day of month in output label	P1	Principal remaining - always decreases
E	End-of-job flag	P2	Total payment made
F	Parameter in function FNR( )	P3	Amount paid to principal
I	Interest due in any payment	P4	Total of all payments
I1	Year-to-date interest	P5	Year-to-date amount to principal
J	Parameter in function FNR( )	P9	Total principal - used in output
K	Index value controlling for next loop	R	Annual interest rate - input as a percentage converted to decimal
M1	Month of year in output label		
M2	Monthly interest (as a decimal)	Y	Year in output label
M9	Counter for number of months	K$	Keyboard input - does not require a return
N	Remaining principal	Q$	Keyboard input - usually requires a return
NO	Number of different payments	X$	Alpha input variables usually converted to numeric to enter data into the program
N1	Payment number to change payments	Y$	
N2	Number of payment changes made so far	Z$	
P(10,2)	P(J,1) is payment number when a change is to occur P(J,2) is amount of the payment		

NOTES AND REFERENCES

Notes:

1. Principal, interest rate have to be positive.
2. Where payment is zero, the term of mortgage cannot be zero.
3. When payment is zero, the program will compute it and display the answer to two decimal digits.
4. When term of mortgage is zero, the program will compute it and display the results.
5. Program will accept any beginning date. The program does not adjust the day of the month to be "reasonable" (Feb. 31 is possible).

References:

SOME COMMON BASIC PROGRAMS, Pool, Lon & Mary Borchers, (Adam Osborne & Assoc., 1978) page 38.

YOUR AMORTIZATION PROGRAM, DeLuca, Charles in THE BEST OF PERSONAL COMPUTING, (Benwill Publishing, 1979) pages 59-61.

PROGRAM LISTING

```
 10 ! Amortize a mortgage,given
 20 ! principal, interest, and
 30 ! the term and the variable
 40 ! payment.
 50 !
 60 ! revision 11/01/82 - twb -
 70 !
 80 ! Single upper-case key in
 90 !
100 DEF FNK$                                  -FNK$ returns a single
                                                uppercase keyboard input
110 K$=KEY$ @ IF K$="" THEN 110
120 FNK$=UPRC$(K$)
130 END DEF
140 ! Round F to J decimals
150 !
160 DEF FNR(F,J)
170 F=INT(F*10^J+.5)/10^J                     -FNR rounds F to J places
180 FNR=F
190 END DEF
200 !
210 DIM P(10,2)
220 Y=0 @ M=0 @ P5=0 @ I1=0 @ P4=0
230 E=0 @ Y9=0 @ M9=0 @ DELAY 2
240 DISP "    $ Mortgage Amortization $   -Sign on message
    "
250 !
260 ! Ask for input data
270 !
280 DISP "Principal"; @ INPUT X$ @ ON E   -Display item to be input and
    RROR GOTO 320                           prepare error routine
290 IF X$="" THEN 280                     -If only RTN is pressed for
                                            principal, repeat prompt
300 IF X$="Q" THEN 1370                   -If input is 'Q' terminate
                                            program
310 P1=VAL(X$) @ P9=P1 @ OFF ERROR @ GO   -Convert alpha to numeric and
    TO 330                                  turn error trap off
320 DISP "Oops..."; @ GOTO 280            -If any error occurred on
                                            input, repeat with error msg.
330 IF P1<=0 THEN 320                     -If the principal is not
                                            positive, repeat input prompt
340 DISP "Annual interest rate (%)"; @
    INPUT X$ @ ON ERROR GOTO 380
350 IF X$="" THEN 340
360 IF X$="Q" THEN 1370
370 R=VAL(X$) @ OFF ERROR @ GOTO 390
380 DISP "Oops.."; @ GOTO 340
390 IF R<=0 THEN 380                      -If the annual interest rate is
                                            zero, repeat input
400 R=R/100 @ M2=R/12                     -Convert percentage to decimal
                                            and compute monthly int.
410 !
420 DISP "Number of different payments"
    ; @ INPUT X$ @ ON ERROR GOTO 460
```

PROGRAM LISTING

```
430 IF X$="" THEN 420
440 IF X$="Q" THEN 1370
450 N0=VAL(X$) @ OFF ERROR @ GOTO 470
460 DISP "Oops..."; @ GOTO 420
470 IF N0<=1 OR N0>10 THEN 460

480 FOR K=1 TO N0

490 DISP "Payment #,payment $"; @ INPUT
     X$,Y$
500 IF X$="" OR Y$="" THEN 490
510 IF X$="Q" THEN 1370
520 ON ERROR GOTO 540
530 P(K,1)=VAL(X$) @ P(K,2)=VAL(Y$) @ O
    FF ERROR @ GOTO 550
540 DISP "Oops..."; @ GOTO 490
550 NEXT K

560 IF P(1,1)#1 THEN 480

570 N1=P(2,1) @ P2=P(1,2) @ N2=1

580 DISP "Beginning date (M,D,Y) "; @ I
    NPUT X$,Y$,Z$ @ ON ERROR GOTO 610
590 IF X$="" OR Y$="" OR Z$="" THEN 580
600 M1=VAL(X$) @ D=VAL(Y$) @ Y=VAL(Z$)
    @ OFF ERROR @ GOTO 620
610 DISP "Oops..."; @ GOTO 580
620 IF Y<0 OR D<0 OR M1<0 THEN 610

630 IF D>31 OR M1>12 THEN 610
640 !
650 ! Compute the table values
660 !
670 INPUT "Do you wish a printout? (Y/N
    )"; X$ @ X$=UPRC$(X$)
680 IF X$='Y' OR X$='N' THEN 690 ELSE 6
    70
690 IF X$='N' THEN 710
700 OFF IO @ RESTORE IO @ DISPLAY IS ":
    pr"
710 DISP TAB(5);"Amortization Table"
720 !
730 M9=M9+1
740 IF N1#M9 THEN 780

750 N2=N2+1 @ P2=P(N2,2)
760 IF N2>=N0 THEN 780

770 N1=P(N2+1,1)
780 I=P1*M2
790 I1=I1+I
800 P3=P2-I
810 P5=P5+P3
```

-If number of payments is invalid, repeat input prompt
-Begin loop to enter payment schedule

-End of entry loop for payment schedule
-If first year is not 1 then repeat schedule entry
-Set up vars for converting payment pattern

-Check validity of beginning date

-Ask for printout option

-Increment month counter
-If month count <> month then the payment changes

-If all payments used, do not advance payment counter

PROGRAM LISTING

```
 820 IF P1<P2 THEN P4=P4+P1 @ GOTO 840      -Handle case where balance due
                                             is less than payment
 830 P4=P4+P3
 840 N=P1-P3
 850 IF N>.005 THEN 890
 860 P5=P5-P3+P1
 870 P3=P1
 880 N=0
 890 IF X$='N' THEN 930 ELSE DISP
 900 DISP TAB(8);STR$(M1)&'/'&STR$(D)&'/
     '&STR$(Y)
 910 DISP " P= ";FNR(P3,2);" I= ";FNR(I,
     2)
 920 GOTO 950
 930 DISP STR$(M1)&"/"&STR$(D)&"/"&STR$(
     Y);" P= ";FNR(P3,2);"I= ";FNR(I,2)
 940 GOSUB 1280 @ IF NUM(Q$)=8 THEN 930
 950 IF N<.005 THEN 1110
 960 IF M1=12 THEN 990                       -If current month is Dec
                                              proceed to year end routines
 970 M1=M1+1                                 -Increment count for month
 980 GOTO 1090
 990 IF X$='N' THEN 1030
1000 DISP @ DISP TAB(8);Y;' End:'
1010 DISP " P= ";FNR(P5,2);' I= ';FNR(I1
     ,2) @ DISP
1020 GOTO 1050
1030 DISP Y;"end P=";FNR(P5,2);"I=";FNR(
     I1,2)
1040 GOSUB 1280 @ IF NUM(Q$)=8 THEN 1030    -If BACK key was pressed
                                             display year end data again
1050 P5=0 @ I1=0                            -Re-initialize yearly totals
1060 IF E=1 THEN DISPLAY IS "*" @ GOTO 1
     340
1070 M1=1
1080 Y=Y+1
1090 P1=N
1100 GOTO 730
1110 IF M1#1 THEN E=1 @ GOTO 990
1120 !
1130 ! View the data
1140 !
1150 DISP "Principal is ";FNR(P9,2)
1160 GOSUB 1280 @ IF NUM(Q$)=8 THEN 1150
1170 DISP "Annual interest (%) ";FNR(R*1
     00,2)
1180 GOSUB 1280 @ IF NUM(Q$)=8 THEN 1150
1190 FOR K=1 TO N0                          -Output payment schedule
1200 DISP "Monthly Payment";K;"=";FNR(P(
     K,2),2)
1210 GOSUB 1280 @ IF NUM(Q$)=13 THEN 123
     0
1220 K=K-2 @ IF K<1 THEN K=0
1230 NEXT K
```

PROGRAM LISTING

```
1240 GOTO 1340
1250 !
1260 ! keyboard input subroutine
1270 !
1280 Q$=FNK$ @ IF NUM(Q$)#13 AND NUM(Q$)
     #8 AND NUM(Q$)#142 THEN 1280
1290 IF NUM(Q$)=142 THEN 1370
1300 RETURN
1310 !
1320 ! Present options menu
1330 !
1340 DISP CHR$(210);"un again,";CHR$(214
     );"iew again, or ";CHR$(197); @ INP
     UT "nd?";Q$
1350 Q$=UPRC$(Q$)
1360 ON POS('RVE',Q$)+1 GOTO 1340,220,11
     50,1370
1370 DELAY 1 @ DISP @ STOP
```

-Moniter keyboard for RTN, BACK, or TAB keys

PROGRAM DESCRIPTION

VARIABLE INTEREST RATE MORTGAGE

The program accepts as input the principal of a mortgage and the fixed payment on that mortgage. The program then asks for the number of variable interest rates, and then asks for the effective payment number of an interest rate and the amount of that interest rate. The date the first payment is due is also requested.

The program will then produce an amortization table for the mortgage. The table may be printed* at the user's choice.

The interest rate will change beginning on the effective month of the rate.

*For a printed table, a printer must be attached and defined as device ":PR".

SAMPLE PROBLEM

Sarah Anne is borrowing $1,000 with a payment of $100 per month. She will pay 15% interest from month 1 to month 5 and will then pay 18% interest. The first payment is on August 10, 1983.

What is her repayment schedule?

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program		
1a	Sign-on message	$ Variable Interest Mortgage $	
2	Enter the principal	Principal?	1000 [RTN]
	Enter the monthly payment	Monthly payment?	100 [RTN]
	Enter the number of different rates	Number of interest rates?	2 [RTN]
	Enter payment number and rate	Payment #, interest rate?	1,15 [RTN]
	Enter payment number and rate	Payment #, interest rate?	6,18 [RTN]
	Enter the beginning date	Beginning date (M,D,Y)?	8,10,1983 [RTN]
3	Display the table	Do you wish a printout? (Y/N)	N [RTN]
4	Amortization table:	Amortization Table	
	Display date, payment to	8/10/1983 P=87.5 I=12.5	[RTN]
	principal and interest	9/10/1983 P=88.59 I=11.41	[RTN]
		10/10/1983 P=89.7 I=10.3	[RTN]
		11/10/1983 P=90.82 I=9.18	[RTN]
		12/10/1983 P=91.96 I=8.04	[RTN]
	Display year-to-date totals	1983 end P=448.57 I=51.43	[RTN]
	Begin the next year	1/10/1984 P=91.73 I=8.27	[RTN]
		2/10/1984 P=93.1 I=6.9	[RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
		3/10/1984 P=94.5 I=5.5	[RTN]
		4/10/1984 P=95.92 I=4.08	[RTN]
		5/10/1984 P=97.36 I=2.64	[RTN]
		6/10/1984 P=78.82 I=1.18	[RTN]
		1984 end P=551.43 I=28.57	[RTN]
3	Review the input data	Run again, View again, or End?	V [RTN]
		Principal is 1000	[RTN]
		Monthly payment 100	[RTN]
		Annual interest 1 = 15	[RTN]
		Annual interest 2 = 18	[RTN]
	End program	Run again, View again, or End?	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	$ Variable Interest Mortgage $	
2	Enter each data item as	Principal?	n [RTN]
	prompted. If "Q" (upper-case	Monthly payment?	p [RTN]
	"Q") is entered, the program	Number of interest rates?	# [RTN]
	will terminate	Payment #, interest rate?	m [RTN]
		(repeat until done)	
		Beginning date (M,D,Y)?	m,d,y [RTN]
		Do you wish a printout? (Y/N)	Y or N [RTN]
3	If 'N' entered: To advance	Amortization Table	
	to next line, press [RTN],	mm/dd/yyyy P=*value* I=*value*	[RTN]/[BACK]
	[TAB]will exit program. [BACK]	(repeat until year end)	
	repeats the current line	yyyy end P=*value* I=*value*	[RTN]/[BACK]
	If 'Y' entered	Amortization Table	
		mm/dd/yyyy	
		P=*value* I=*value*	
		(repeat until year end)	
		yyyy end P=*value* I=*value*	
		(repeat until done)	
4	Display options menu	Run again, View again, or End?	R,V, or E [RTN]
5	If 'R' is pressed goto 1		
	If 'V' is pressed	Principal is	[RTN]/[BACK]
	[RTN] displays next item	Monthly payment	[RTN]/[BACK]
	[BACK] display prior item	Annual interest	[RTN]/[BACK]
	[TAB] ends the program	(repeat until end)	
	If 'E' is pressed end program.		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
D	Day of month in-output label	P(10,2)	P(J,1) is payment number when a change is to occur P(J,2) is amount of the interest rate
E	End-of-job flag		
F	Parameter in function FNR( )	P1	Principal remaining - always decreases
I	Interest due on any payment	P2	Total payment made
I1	Year-to-date interest	P3	Amount paid to principal
J	Parameter in function FNR( )	P4	Total of all payments
K	Index value controlling for next loop	P5	Year-to-date amount to principal
M1	Month of year in-output label	P9	Total principal - used in output
M2	Monthly interest (as a decimal)	R	Annual interest rate - input as a percentage and converted to decimal
M9	Counter for number of months		
N	Remaining principal	Y	Year in output label
NO	Number of different interest rates	K$	Keyboard input - does not require a return
N1	Payment number to change interest rates	Q$	Keyboard input - usually requires a return
N2	Number of interest changes made so far	X$	Alpha input variables usually converted to numeric to enter data into the program
		Y$	
		Z$	

NOTES AND REFERENCES

References:

1. Principal, interest, and payment must always be positive.

2. There may be no more than 10 different interest rates.

3. The month numbers have to increase. No month number may be less than the month number before it.

YOUR AMORTIZATION PROGRAM, DeLuca, Charles, in THE BEST OF PERSONAL COMPUTING, (Benwill Publishing, 1979) pp. 59-61.

PROGRAM LISTING

```
 10 ! Amortize a mortgage, given
 20 ! principal, payment and
 30 ! the term and the variable
 40 ! interest.
 50 !
 60 ! revision 11/01/82 - twb -
 70 !
 80 !
 90 ! single upper-case key in
100 !
110 DEF FNK$                                  -FNK$ returns a single
                                               uppercase keyboard input
120 K$=KEY$ @ IF K$="" THEN 120
130 FNK$=UPRC$(K$)
140 END DEF
150 !
160 ! Round F to J decimals
170 !
180 DEF FNR(F,J)                              -FNR rounds F to J digits
190 F=INT(F*10^J+.5)/10^J
200 FNR=F
210 END DEF
220 !
230 DIM P(10,2)
240 Y=0 @ M=0 @ P5=0 @ I1=0 @ P4=0
250 E=0 @ Y9=0 @ M9=0 @ DELAY 2
260 DISP " $ Variable Interest Mortgage       -Sign on message
     $"
270 !
280 ! Ask for input data
290 !
300 DISP "Principal"; @ INPUT X$ @ ON E       -Display item to be entered and
    RROR GOTO 340                              prepare error routine
310 IF X$="" THEN 300                         -If null input ask again
320 IF X$="Q" THEN 1400                       -If 'Q' then terminate program
330 P1=VAL(X$) @ P9=P1 @ OFF ERROR @ GO       -Convert alpha value to numeric
    TO 350                                     and end error trap
340 DISP "Oops..."; @ GOTO 300                -If error occurred display
                                               error message and ask again
350 IF P1<=0 THEN 340                         -If principal is not positive
                                               ask again
360 DISP "Monthly payment"; @ INPUT X$
    @ ON ERROR GOTO 400
370 IF X$="" THEN 360
380 IF X$="Q" THEN 1400
390 P2=VAL(X$) @ OFF ERROR @ GOTO 410
400 DISP "Oops..."; @ GOTO 360
410 IF P2<=0 THEN 400                         -If the payment is not positive
                                               ask again
420 !
430 DISP "Number of interest rates"; @
    INPUT X$ @ ON ERROR GOTO 470
440 IF X$="" THEN 430
450 IF X$="Q" THEN 1400
```

PROGRAM LISTING

```
460 N0=VAL(X$) @ OFF ERROR @ GOTO 480
470 DISP "Oops..."; @ GOTO 430
480 IF N0<=1 OR N0>10 THEN 470

490 FOR K=1 TO N0

500 DISP "Payment #,interest rate"; @ I
    NPUT X$,Y$
510 IF X$="" OR Y$="" THEN 500
520 IF X$="Q" THEN 1400
530 ON ERROR GOTO 560
540 P(K,1)=VAL(X$) @ P(K,2)=VAL(Y$) @ O
    FF ERROR
550 IF P(K,2)>0 THEN 570
560 DISP "Oops..."; @ GOTO 500
570 NEXT K

580 IF P(1,1)#1 THEN 490
590 N1=P(2,1) @ R=P(1,2) @ N2=1

600 R=R/100 @ M2=R/12

610 DISP "Beginning date (M,D,Y) "; @ I
    NPUT X$,Y$,Z$ @ ON ERROR GOTO 640
620 IF X$="" OR Y$="" OR Z$="" THEN 610
630 M1=VAL(X$) @ D=VAL(Y$) @ Y=VAL(Z$)
    @ OFF ERROR @ GOTO 650
640 DISP "Oops..."; @ GOTO 610
650 IF Y<0 OR D<0 OR M1<0 THEN 640

660 IF D>31 OR M1>12 THEN 640
670 INPUT "Do you wish a printout? (Y/N
    ) "; X$ @ X$=UPRC$(X$)
680 IF X$="Y" OR X$="N" THEN 690 ELSE 6
    70
690 IF X$="N" THEN 750
700 OFF IO @ RESTORE IO
710 DISPLAY IS ":pr"
720 !
730 ! Compute the table values
740 !
750 DISP TAB(5);"Amortization Table"

760 !
770 M9=M9+1
780 IF N1#M9 THEN 820
790 N2=N2+1 @ R=P(N2,2) @ R=R/100 @ M2=
    R/12
800 IF N2>=N0 THEN 820
810 N1=P(N2+1,1)
820 I=P1*M2
830 I1=I1+I
840 P3=P2-I
850 P5=P5+P3
```

-If # of interest rates invalid, ask again

-Begin loop to input the different interest rates

-End the interest rate input loop

-Check for valid first rate

-Set holding variables to convert payment schedule

-Convert interest rate to decimal, compute monthly rate

-Check validity of beginning date

-Indicate that table is to follow

-Increment month counter.

PROGRAM LISTING

```
 860 IF P1<P2 THEN P4=P4+P1 @ GOTO 880
 870 P4=P4+P3
 880 N=P1-P3
 890 IF N>.005 THEN 920
 900 P5=P5-P3+P1 @ P3=P1
 910 N=0
 920 IF X$="N" THEN 960 ELSE DISP
 930 DISP TAB(10);STR$(M1)&"/"&STR$(D)&"
     /"&STR$(Y)
 940 DISP "P= ";FNR(P3,2);" I= ";FNR(I,2
     )
 950 GOTO 980
 960 DISP STR$(M1)&"/"&STR$(D)&"/"&STR$(
     Y);" P= ";FNR(P3,2);"I= ";FNR(I,2)
 970 GOSUB 1310 @ IF NUM(Q$)=8 THEN 960
 980 IF N<.005 THEN 1140
 990 IF M1=12 THEN 1020
1000 M1=M1+1
1010 GOTO 1120
1020 IF X$="N" THEN 1060
1030 DISP @ DISP TAB(10);Y;" end:"
1040 DISP " P= ";FNR(P5,2);" I= ";FNR(I1
     ,2) @ DISP
1050 GOTO 1080
1060 DISP Y;"end P=";FNR(P5,2);"I= ";FNR
     (I1,2)
1070 GOSUB 1310 @ IF NUM(Q$)=8 THEN 1020
1080 P5=0 @ I1=0
1090 IF E=1 THEN DISPLAY IS "*" @ GOTO 1
     370
1100 M1=1
1110 Y=Y+1
1120 P1=FNR(N,2)
1130 GOTO 770
1140 IF M1#1 THEN E=1 @ GOTO 1020
1150 !
1160 ! View the data
1170 !
1180 DISP "Principal is ";FNR(P9,2)
1190 GOSUB 1310 @ IF NUM(Q$)=8 THEN 1180
1200 DISP "Monthly payment";FNR(P2,2)
1210 GOSUB 1310 @ IF NUM(Q$)=8 THEN 1180
1220 FOR K=1 TO N0
1230 DISP "Annual Interest";K;"=";FNR(P(
     K,2),2)
1240 GOSUB 1310 @ IF NUM(Q$)=13 THEN 126
     0
1250 K=K-2 @ IF K<1 THEN K=0
1260 NEXT K
1270 GOTO 1370
1280 !
1290 ! keyboard input subroutine
1300 !
1310 Q$=FNK$ @ IF NUM(Q$)#13 AND NUM(Q$)
     #8 AND NUM(Q$)#142 THEN 1310
```

PROGRAM LISTING

```
1320 IF NUM(Q$)=142 THEN 1400
1330 RETURN
1340 !
1350 ! Present options menu
1360 !
1370 DISP CHR$(210);"un again,";CHR$(214
     );"iew again, or ";CHR$(197); @ INP
     UT "nd?";Q$
1380 Q$=UPRC$(Q$)
1390 ON POS('RVE',Q$)+1 GOTO 1370,240,11
     80,1400
1400 DELAY 1 @ DISP @ STOP
```

PROGRAM DESCRIPTION

LOAN SCHEDULE

This program calculates monthly payments when the user knows the principal, annual interest rate, and term for a loan. The program will also calculate the term for a loan, given the principal, annual interest rate, and monthly payment.

The program will produce an amortization table listing the payments toward principal and interest for each month, as well as annual totals. The loan may start on any date.

Thus, the user may know the term of a loan, and use this program to compute the monthly payments, or use this program to compute the term of a loan given a known monthly payment.

Amortized loans, where total payment = payment to principal and payment to interest, can be handled by this program. It is possible to handle interest-only loans, where total payment = payment to interest, and no payment is made to principal.

If a printer is attached and defined as device ":PR", the user has the option available to print the amortization table.

SAMPLE PROBLEM

Bob Edwards wishes to borrow $1,500 from his father. They have negotiated an interest rate of 18.73 percent. If Bob means to repay the loan in one year, two months, what will monthly payments be?

Answer: $120.11

What is the repayment schedule if the first payment is to be made on September 13, 1983?

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program		
	Sign-on message	$ Mortgage Amortization $	
	Enter principal of loan	Principal?	1500 [RTN]
	Enter annual interest rate	Annual interest rate (%)?	18.73 [RTN]
	Solve for monthly payment	Monthly payment?	0 [RTN]
	Enter term in years and months	Term in years, months?	1,2 [RTN]
	Display monthly payment	Monthly payment is 120.11	[RTN]
	Enter beginning date	Beginning date (M,D,Y)?	9,13,1983 [RTN]
	Ask if printout is wanted	Do you wish a printout? (Y/N)	N [RTN]
2	Display results	Amortization Table	
	Show date, principal and	9/13/1983 P=96.69 I=23.41	[RTN]
	interest	10/13/1983 P=98.2 I=21.9	[RTN]
		11/13/1983 P=99.74 I=20.37	[RTN]
		12/13/1983 P=101.29 I=18.81	[RTN]
		1983 end: P=395.92 I=84.5	[RTN]
		1/13/1984 P=102.87 I=17.23	[RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
		2/13/1984 P=104.48 I=15.63	[RTN]
		3/13/1984 P=106.11 I=14	[RTN]
		4/13/1984 P=107.77 I=12.34	[RTN]
		5/13/1984 P=109.45 I=10.66	[RTN]
		6/13/1984 P=111.16 I= 8.95	[RTN]
		7/13/1984 P=112.89 I= 7.21	[RTN]
		8/13/1984 P=114.65 I= 5.45	[RTN]
		9/13/1984 P=116.44 I= 3.66	[RTN]
		10/13/1984 P=118.26 I= 1.85	[RTN]
		1984 end: P=1104.08 I=96.98	[RTN]
3	Options menu	Run again, View again, or End?	V [RTN]
4	View input data	Principal is 1500	[RTN]
		Annual interest (%) 18.73	[RTN]
		Monthly payment 120.11	[RTN]
		Term is 1 year and 2 months	[RTN]
5	End program	Run again, View again, or End?	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program		
1a	Sign on message	$ Mortgage Amortization $	
2	Enter data items and press	Principal?	p [RTN]
	[RTN]. To quit at this point	Annual interest rate (%)	i [RTN]
	[RTN].	Monthly payment?	n [RTN]
		Beginning date (M,D,Y)?	m,d,y [RTN]
3	Select output option	Do you wish a printout? (Y/N)	Y or N [RTN]
4a	If 'N' is entered, display	Amortization Table	
	answers at year end display	mm/dd/yyyy P=*value* I=*value*	[RTN]/[BACK]
	[RTN] goes to next item, [BACK]	yyyy end P=*value* I=*value*	[RTN]/[BACK]
	does not change display and		
	[TAB] ends the program		
4b	If 'Y' is entered for step 3	Amortization Table	[RTN]
	print the table	mm/dd/yyyy	[RTN]
		P=*value* I=*value*	[RTN]
		yyyy end	[RTN]
		P=*value* I=*value*	
5	Display options menu	Run again, View again, or End?	R,V or E [RTN]
	If 'V' is selected display	Principal	[RTN]/[BACK]
	[RTN] goes to next item,	Annual interest %	[RTN]/[BACK]
	[BACK] goes to prior item and	Monthly payment	[RTN]/[BACK]
	[TAB] ends the program	Term is __ years, __ months	[RTN]/[BACK]
	Goto 5		
	If 'R' is selected goto 1a		
	If 'E' is selected program ends		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
D	Day of the month of payment	P5	Accumulated payment to principal
E	Flag to indicate report is at an end	P9	Output value for principal (=initial P1)
F	Parameter used on rounding [RTN]	R	Annual interest rate as decimal fraction
I	The interest component of mortgage	S1	Temporary value used to compute years of mort.
I1	Interest accumulated to date (yearly)	S2	Temporary value
J	Precision of rounding routine	S3	Temporary value
M	Monthly component of term of mortgage	T	Term of mortgage in months
M0	Total months of term of mortgage	T1	Term of mortgage in years
M1	Month of payment	Y	Year report begins & current year
M2	Monthly interest rate	Y0	Year component of term of mortgage
N	Principal remaining on mortgage	Y9	Term of mortgage in years
P1	Principal	K$	Keyboard response variable used in functions FNK$
P2	Monthly payment	Q$	Keyboard input variable
P3	Payment to principal	X$	General purpose
P4	Accumulated total payment (yearly)	Y$	input variables,
		Z$	converted to numeric

NOTES AND REFERENCES

Notes:

1. Principal & interest have to be positive.
2. There may be no more than 10 different payments.
3. The payment numbers have to increase. No payment number may be less than the payment number before it.
4. If a printout is desired, then a printer must be defined as device ":PR".

Reference: YOUR AMORTIZATION PROGRAM, DeLuca, Charles in THE BEST OF PERSONAL COMPUTING, (Benwill Publishing, 1979), pp. 59-61.

PROGRAM LISTING

```
 10 ! Amortize a mortgage, given
 20 ! principal, interest, and
 30 ! either the term or the
 40 ! payment, or both.
 50 !
 60 ! Revision 11/01/82 - twb -
 70 !
 80 ! Single upper-case key in
 90 !
100 DEF FNK$                                  -Single uppercase keyboard
                                                input
110 K$=KEY$ @ IF K$="" THEN 110
120 FNK$=UPRC$(K$)
130 END DEF
140 !
150 ! Round F to J decimals
160 !
170 DEF FNR(F,J)                              -Round F to J digits
180 F=INT(F*10^J+.5)/10^J
190 FNR=F
200 END DEF
210 !
220 Y=0 @ M=0 @ P5=0 @ I1=0 @ P4=0            -Zero accumulators and set
                                                delay time to 2 seconds
230 E=0 @ Y9=0 @ DELAY 2
240 DISP "    $ Mortgage Amortization $       -Display sign on message
    "
250 !
260 ! Ask for input data
270 !
280 DISP "Principal"; @ INPUT X$ @ ON E       -Accept value for input,
    RROR GOTO 320                               prepare error trap
290 IF X$="" THEN 280                         -If input is null, ask again
300 IF X$="Q" THEN 1570                       -Entry of 'Q' will terminate
                                                program
310 P1=VAL(X$) @ P9=P1 @ OFF ERROR @ GO       -Convert to numeric, stop error
    TO 330                                      trap
320 DISP "Oops..."; @ GOTO 280                -If input is unacceptable
                                                display warning and ask again
330 IF P1<=0 THEN 320                         -If payment is not positive,
                                                ask again
340 DISP "Annual interest rate (%)"; @
    INPUT X$ @ ON ERROR GOTO 380
350 IF X$="" THEN 340
360 IF X$="Q" THEN 1570
370 R=VAL(X$) @ OFF ERROR @ GOTO 390
380 DISP "Oops..."; @ GOTO 340
390 IF R<=0 THEN 380                          -If interest is not positive,
                                                ask again
400 R=R/100 @ M2=R/12                         -Convert interest to decimal,
                                                convert to monthly rate
410 DISP "Monthly Payment "; @ INPUT X$
     @ ON ERROR GOTO 450
420 IF X$="" THEN 410
```

PROGRAM LISTING

```
430 IF X$="Q" THEN 1570
440 P2=VAL(X$) @ OFF ERROR @ GOTO 460
450 DISP "Oops..."; @ GOTO 410
460 IF P2<0 THEN 450                          -If payment is negative, ask
                                               again
470 DISP "Term in years,months "; @ INP
    UT X$,Y$ @ ON ERROR GOTO 510
480 IF X$="" OR Y$="" THEN 470
490 IF X$="Q" THEN 1570
500 Y0=VAL(X$) @ M=VAL(Y$) @ OFF ERROR
    @ GOTO 520
510 DISP "Oops..."; @ GOTO 470
520 IF Y0<0 OR M<0 THEN 510
530 M0=Y0*12+M                                -Compute the total number of
                                               months in the term
540 IF P2=0 AND Y0=0 AND M=0 THEN 450         -Trap invalid term
550 IF P2=0 THEN 650
560 IF Y0=0 AND M=0 THEN 750
570 T1=Y0 @ T=M @ GOTO 840
580 !
590 ! The monthly payment is
600 ! not known, so solve for
610 ! it, using then following
620 ! formula:
630 ! p2=p1*(m2*(1+m2)^m0)/((1+m2)^m0-1
    )
640 !
650 P2=P1*(M2*(1+M2)^M0)/((1+M2)^M0-1)
660 DISP "Monthly Payment is ";FNR(P2,2   -Display computed monthly
    ) @ GOSUB 1480 @ IF NUM(Q$)=8 THEN     payment
    660
670 GOTO 840
680 !
690 ! Term is unknown, sosolve
700 ! for it. Use the following
710 ! formula:
720 !
730 ! y9=-(log(1-p1*r/12*p2)/(log(1-m2)
    *12)
740 !
750 S1=P1*R
760 S2=12*P2
770 S3=1+M2
780 Y9=-LOG(1-S1/S2)/(LOG(S3)*12)
790 T=INT(Y9*12+.5)
800 T1=INT(T/12)
810 T=T-T1*12
820 DISP "Term is ";T1;"years";T;"month   -Display term of the loan
    s" @ GOSUB 1480 @ IF NUM(Q$)=8 THEN
     820
830 Y0=T1 @ M=T
840 DISP "Beginning date (M,D,Y) "; @ I
    NPUT X$,Y$,Z$ @ ON ERROR GOTO 870
850 IF X$="" OR Y$="" OR Z$="" THEN 840
```

PROGRAM LISTING

```
 860 M1=VAL(X$) @ D=VAL(Y$) @ Y=VAL(Z$)
     @ OFF ERROR @ GOTO 880
 870 DISP "Oops..."; @ GOTO 840
 880 IF Y<0 OR D<0 OR M1<0 THEN 870          -Check for validity of date
 890 IF D>31 OR M1>12 THEN 870
 900 !
 910 ! Compute the table values
 920 !
 930 INPUT "Do you wish a printout? (Y/N     -Ask for output option
     )"; X$ @ X$=UPRC$(X$)
 940 IF X$='Y' OR X$='N' THEN 950 ELSE 9
     30
 950 IF X$='N' THEN 970
 960 OFF IO @ RESTORE IO @ DISPLAY IS ":
     PR"
 970 DISP TAB(5);"Amortization Table"
 980 !
 990 I=P1*M2
1000 I1=I1+I                                 -Accumulate interest paid
1010 P3=P2-I
1020 P5=P5+P3                                -Accumulate principal paid
1030 IF P1<P2 THEN P4=P4+P1 @ GOTO 1050      -If pmt > balance accumulate
                                              amount paid to principal
1040 P4=P4+P3                                -Otherwise accumulate amount
                                              paid to total principal
1050 N=P1-P3                                 -Compute remaining balance
                                              after pmt to principal
1060 IF N>.005 THEN 1100
1070 P5=P5-P3+P1
1080 P3=P1
1090 N=0
1100 IF X$='N' THEN 1140
1110 DISP TAB(8);STR$(M1)&'/'&STR$(D)&'/
     '&STR$(Y)
1120 DISP ' P= ';FNR(P3,2);' I= ';FNR(I,
     2) @ DISP
1130 GOTO 1160
1140 DISP STR$(M1)&"/"&STR$(D)&"/"&STR$(     -Display data showing monthly
     Y);" P= ";FNR(P3,2);"I= ";FNR(I,2)       figures
1150 GOSUB 1480 @ IF NUM(Q$)=8 THEN 1140
1160 IF N<.005 THEN GOTO 1320                -If balance is less than a
                                              half-cent, prepare for end
1170 IF M1=12 THEN 1200                      -If current month is December,
                                              goto year end report
1180 M1=M1+1                                 -Increment month counter
1190 GOTO 1300                               -Bypass year end reports
1200 IF X$='N' THEN 1240
1210 DISP @ DISP TAB(8);Y;' End:'
1220 DISP ' P= ';FNR(P5,2);' I= ';FNR(I1
     ,2) @ DISP @ DISP @ DISP
1230 GOTO 1260
1240 DISP Y;"end P=";FNR(P5,2);"I=";FNR(     -Show year end totals
     I1,2)
1250 GOSUB 1480 @ IF NUM(Q$)=8 THEN 1240
```

PROGRAM LISTING

```
1260 P5=0 @ I1=0                          -Zero out yearly accumulators
1270 IF E=1 THEN DISPLAY IS "*" @ GOTO 1
     540
1280 M1=1                                 -Set month counter to January
1290 Y=Y+1                                -Increment year counter
1300 P1=N                                 -Update remaining balance
1310 GOTO 990                             -Return to compute another
                                           month's data
1320 IF M1$1 THEN E=1 @ GOTO 1200
1330 !
1340 ! View the data
1350 !
1360 DISP "Principal is ";FNR(P9,2)       -Display principal
1370 GOSUB 1480 @ IF NUM(Q$)=8 THEN 1360
1380 DISP "Annual interest (%) ";FNR(R*1  -Display interest rate
     00,2)
1390 GOSUB 1480 @ IF NUM(Q$)=8 THEN 1360
1400 DISP "Monthly Payment ";FNR(P2,2)    -Display monthly payment
1410 GOSUB 1480 @ IF NUM(Q$)=8 THEN 1380
1420 DISP "Term is";Y0;"years";M;"months  -Display term in years and
     "                                     months
1430 GOSUB 1480 @ IF NUM(Q$)=8 THEN 1400
1440 GOTO 1540
1450 !
1460 ! Keyboard input subroutine
1470 !
1480 Q$=FNK$ @ IF NUM(Q$)$13 AND NUM(Q$)  -Monitor keyboard for RTN,
     $8 AND NUM(Q$)$142 THEN 1480          BACK, or TAB keys.
1490 IF NUM(Q$)=142 THEN 1570
1500 RETURN
1510 !
1520 ! Present options menu
1530 !
1540 DISP CHR$(210);"un again,";CHR$(214  -Continuation options
     );"iew again, or ";CHR$(197); @ INP
     UT "nd?";Q$
1550 Q$=UPRC$(Q$)
1560 ON POS('RVE',Q$)+1 GOTO 1540,220,13
     60,1570
1570 DELAY 1 @ DISP @ STOP
```

NOTES

NOTES

NOTES

NOTES

REAL ESTATE

INCOME PROPERTY ANALYSIS
ESTIMATE OF BUYER'S COSTS
SELLER'S COSTS AND NET EQUITY
INTERNAL RATE OF RETURN
RENT VS. BUY
VARIABLE PAYMENT MORTGAGE AMORTIZATION TABLES
VARIABLE INTEREST RATE MORTGAGE
LOAN SCHEDULE

ALL HP-75 SOLUTIONS BOOKS ARE AVAILABLE RECORDED ON MINI-DATA CASSETTES FROM EITHER A HEWLETT-PACKARD DEALER OR THE HP USERS' LIBRARY.

00075-90070 Printed in U.S.A.

Reorder No.
00075-13010